Multi-Parameter Laser Modes in Paraxial Optics

© Christoph Koutschan, Erwin Suazo, and Sergei K. Suslov

(Last modified on 20 September 2015.)

The HolonomicFunctions package can be downloaded from
http://www.risc.jku.at/research/combinat/software/HolonomicFunctions/

In[1]:=
```
<<RISC`HolonomicFunctions`
```

HolonomicFunctions Package version 1.7.1 (09-Oct-2013)
written by Christoph Koutschan


```
--> Type  ?HolonomicFunctions  for help.
```

2.1.1 Green's Function and Generalized Fresnel Integrals

In[2]:=
```
(* This is Equation (2.1). *)
schroedinger= I*D[ψ[x, t], t] + a[t]*D[ψ[x, t], x, x] +
    (I*x*c[t] - I*g[t])*D[ψ[x, t], x] + (-x^2*b[t] + I*d[t] + x*f[t])*ψ[x, t]
```

Out[2]=
```
(-x² b[t] + ⅈ d[t] + x f[t]) ψ[x, t] + ⅈ ψ^(0,1)[x, t] + (ⅈ x c[t] - ⅈ g[t]) ψ^(1,0)[x, t] + a[t] ψ^(2,0)[x, t]
```

Derive the differential equations for α, ..., κ, starting from the Schrödinger equation, using the Green's function. We end up with a Riccati-type system, compare Equations (2.41) - (2.47) with $c_0 = 0$.

```
(* Plug in the Green's function G(x,y,t) given in (2.3). *)
(* Divide by the exponential term . *)
Numerator [Together[
    With[{exp = Exp[I*(α[t]*x^2 + δ[t]*x + κ[t] + β[t]*x*y + ϵ[t]*y + γ[t]*y^2)]},
      (schroedinger/. ψ → Function@@{{x, t}, 1/Sqrt[2 Pi*μ[t]]*exp}) / exp / Sqrt[2]]]]
```

```
-2 x² b[t] μ[t] + 2 ⅈ d[t] μ[t] + 2 x f[t] μ[t] + 4 ⅈ a[t] α[t] μ[t] - 4 x² c[t] α[t] μ[t] +
  4 x g[t] α[t] μ[t] - 8 x² a[t] α[t]² μ[t] - 2 x y c[t] β[t] μ[t] + 2 y g[t] β[t] μ[t] -
  8 x y a[t] α[t] β[t] μ[t] - 2 y² a[t] β[t]² μ[t] - 2 x c[t] δ[t] μ[t] + 2 g[t] δ[t] μ[t] -
  8 x a[t] α[t] δ[t] μ[t] - 4 y a[t] β[t] δ[t] μ[t] - 2 a[t] δ[t]² μ[t] - 2 x² μ[t] α'[t] -
  2 x y μ[t] β'[t] - 2 y² μ[t] γ'[t] - 2 x μ[t] δ'[t] - 2 y μ[t] ϵ'[t] - 2 μ[t] κ'[t] - ⅈ μ'[t]
```

```
(* Coefficient comparison w.r.t. x, y, and the imaginary unit. *)
riccati= DeleteCases[Flatten[CoefficientList[% /. Complex [a_, b_] :> a + i*b, {x, y, i}]], 0]
```

```
{2 g[t] δ[t] μ[t] - 2 a[t] δ[t]² μ[t] - 2 μ[t] κ'[t], 2 d[t] μ[t] + 4 a[t] α[t] μ[t] - μ'[t],
  2 g[t] β[t] μ[t] - 4 a[t] β[t] δ[t] μ[t] - 2 μ[t] ϵ'[t], -2 a[t] β[t]² μ[t] - 2 μ[t] γ'[t],
  2 f[t] μ[t] + 4 g[t] α[t] μ[t] - 2 c[t] δ[t] μ[t] - 8 a[t] α[t] δ[t] μ[t] - 2 μ[t] δ'[t],
  -2 c[t] β[t] μ[t] - 8 a[t] α[t] β[t] μ[t] - 2 μ[t] β'[t],
  -2 b[t] μ[t] - 4 c[t] α[t] μ[t] - 8 a[t] α[t]² μ[t] - 2 μ[t] α'[t]}
```

```
(* This gives the Riccati-type system . *)
riccati =
  Sort[Solve[# == 0, First[Cases[#, Derivative[_][_][_], Infinity]]][[1, 1]] & /@ riccati];
TableForm [riccati /. a1_[t] → a1]
```

$\alpha' \to -b - 2 c \alpha - 4 a \alpha^2$
$\beta' \to -c \beta - 4 a \alpha \beta$
$\gamma' \to -a \beta^2$
$\delta' \to f + 2 g \alpha - c \delta - 4 a \alpha \delta$
$\epsilon' \to g \beta - 2 a \beta \delta$
$\kappa' \to g \delta - a \delta^2$
$\mu' \to 2 (d \mu + 2 a \alpha \mu)$

In[3]:=
```
(* For initialization only. *)
riccati = {α'[t] → -b[t] - 2 c[t] α[t] - 4 a[t] α[t] ^ 2,
     β'[t] → -c[t] β[t] - 4 a[t] α[t] β[t], γ'[t] → -a[t] β[t] ^ 2,
     δ'[t] → f[t] + 2 g[t] α[t] - c[t] δ[t] - 4 a[t] α[t] δ[t], ϵ'[t] → g[t] β[t] - 2 a[t] β[t] δ[t],
     κ'[t] → g[t] δ[t] - a[t] δ[t] ^ 2, μ'[t] → 2 (d[t] μ[t] + 2 a[t] α[t] μ[t])};
```

In[4]:=
```
(* This is the fundamental  solution (2.4) - (2.9) of the Riccati-type system . *)
eqns24to29 = {
     α0 → Function[t, (4 * a[t]) ^ (-1) * D[μ0[t], t] / μ0[t] - d[t] / (2 * a[t])],
     β0 → Function[t, -λ[t] / μ0[t]],
     λ → Function[t, Exp[-Integrate[c[ζ] - 2 * d[ζ], {ζ, 0, t}]]],
     γ0 → Function[t, (2 * μ1[0]) ^ (-1) * μ1[t] / μ0[t] + d[0] / (2 * a[0])],
     δ0 → Function[t, λ[t] / μ0[t] * Integrate[
            ((f[ξ] - d[ξ] / a[ξ] * g[ξ]) * μ0[ξ] + g[ξ] / (2 * a[ξ]) * μ0 '[ξ]) / λ[ξ], {ξ, 0, t}]],
     ϵ0 → Function[t, -2 * a[t] * λ[t] * δ0[t] / μ0 '[t] +
          8 * Integrate[a[s] * σ[s] * λ[s] * μ0[s] * δ0[s] / μ0 '[s] ^ 2, {s, 0, t}]
          + 2 * Integrate[a[s] * λ[s] / μ0 '[s] * (f[s] - d[s] / a[s] * g[s]), {s, 0, t}]],
     κ0 → Function[t, a[t] * μ0[t] / μ0 '[t] * δ0[t] ^ 2 -
          4 * Integrate[a[s] * σ[s] / μ0 '[s] ^ 2 * (μ0[s] * δ0[s]) ^ 2, {s, 0, t}] -
          2 * Integrate[a[s] / μ0 '[s] * μ0[s] * δ0[s] * (f[s] - d[s] / a[s] * g[s]), {s, 0, t}]]};
TableForm [(#1[t] == #2[t] & @@@ eqns24to29)] // TraditionalForm
```

Out[5]//TraditionalForm=

$$\alpha_0(t) = \frac{\mu_0'(t)}{4\, a(t)\, \mu_0(t)} - \frac{d(t)}{2\, a(t)}$$

$$\beta_0(t) = -\frac{\lambda(t)}{\mu_0(t)}$$

$$\lambda(t) = e^{-\int_0^t (c(\zeta) - 2\, d(\zeta))\, d\zeta}$$

$$\gamma_0(t) = \frac{d(0)}{2\, a(0)} + \frac{\mu_1(t)}{2\, \mu_1(0)\, \mu_0(t)}$$

$$\delta_0(t) = \frac{\lambda(t) \left(\int_0^t \frac{\mu_0(\xi) \left(f(\xi) - \frac{d(\xi)\, g(\xi)}{a(\xi)}\right) + \frac{g(\xi)\, \mu_0'(\xi)}{2\, a(\xi)}}{\lambda(\xi)}\, d\xi \right)}{\mu_0(t)}$$

$$\epsilon_0(t) = 2 \int_0^t \frac{a(s)\, \lambda(s) \left(f(s) - \frac{d(s)\, g(s)}{a(s)}\right)}{\mu_0'(s)}\, ds + 8 \int_0^t \frac{a(s)\, \delta_0(s)\, \lambda(s)\, \mu_0(s)\, \sigma(s)}{\mu_0'(s)^2}\, ds - \frac{2\, a(t)\, \delta_0(t)\, \lambda(t)}{\mu_0'(t)}$$

$$\kappa_0(t) = -2 \int_0^t \frac{a(s)\, \delta_0(s)\, \mu_0(s) \left(f(s) - \frac{d(s)\, g(s)}{a(s)}\right)}{\mu_0'(s)}\, ds - 4 \int_0^t \frac{a(s)\, \delta_0(s)^2\, \mu_0(s)^2\, \sigma(s)}{\mu_0'(s)^2}\, ds + \frac{a(t)\, \delta_0(t)^2\, \mu_0(t)}{\mu_0'(t)}$$

Now we derive the second-order ODE for μ.

```
(* Start with the Riccati equation for alpha, and substitute (2.4). *)
Together[#1 - #2 & @@ riccati[[1]] /. (eqns24to29[[1]] /. Subscript[x_, 0] → x)]
```

$$\frac{1}{4\, a[t]^2 \mu[t]} \left(4\, a[t]^2 b[t]\, \mu[t] - 4\, a[t]\, c[t]\, d[t]\, \mu[t] + 4\, a[t]\, d[t]^2 \mu[t] + 2\, d[t]\, \mu[t]\, a'[t] - 2\, a[t]\, \mu[t]\, d'[t] + 2\, a[t]\, c[t]\, \mu'[t] - 4\, a[t]\, d[t]\, \mu'[t] - a'[t]\, \mu'[t] + a[t]\, \mu''[t]\right)$$

```
(* We obtain precisely Equation (2.10). *)
(eqn210 = Collect[% / Coefficient[%, μ''[t]], μ[t] | μ'[t], Simplify]) /. x_[t] → x
```

$$\mu\left(4\,a\,b - 4\,c\,d + 4\,d^2 + \frac{2\,d\,a'}{a} - 2\,d'\right) + \left(2\,c - 4\,d - \frac{a'}{a}\right)\mu' + \mu''$$

Store some relations for the function μ; the coefficients of its ODE are denoted σ and τ, see Equations (2.10) and (2.11).

```
eqns210to211 = {
    σ[t_] → Coefficient[eqn210, μ[t]] / 4,
    τ[t_] → -Coefficient[eqn210, μ'[t]],
    (* ODE (16) for mu . *)
    Derivative[2][μ : (μ0 | μ1)][t_] → τ[t] * μ'[t] - 4 * σ[t] * μ[t],
    (* Use Wronskian of mu0 and mu1 . *)
    Derivative[1][μ1][t] → (μ1[t] μ0'[t] - 2 a[t] μ1[0] * λ[t]^2) / μ0[t]};
```

In[6]:=
```
(* For initialization only. *)
eqns210to211 =
  {σ[t_] → 1/4 (4 a[t] b[t] - 4 c[t] d[t] + 4 d[t]^2 + (2 d[t] a'[t])/a[t] - 2 d'[t]), τ[t_] → -2 c[t] + 4 d[t] + a'[t]/a[t],
   (μ : μ0 | μ1)''[t_] → -4 μ[t] σ[t] + τ[t] μ'[t], μ1'[t] → 1/μ0[t] (-2 a[t] λ[t]^2 μ1[0] + μ1[t] μ0'[t])}
```

Out[6]=

$$\left\{\sigma[t\_] \to \frac{1}{4}\left(4\,a[t]\,b[t] - 4\,c[t]\,d[t] + 4\,d[t]^2 + \frac{2\,d[t]\,a'[t]}{a[t]} - 2\,d'[t]\right),\ \tau[t\_] \to -2\,c[t] + 4\,d[t] + \frac{a'[t]}{a[t]},\right.$$
$$\left.(\mu : \mu_0 \mid \mu_1)''[t\_] \to -4\,\mu[t]\,\sigma[t] + \tau[t]\,\mu'[t],\ \mu_1'[t] \to \frac{-2\,a[t]\,\lambda[t]^2\,\mu_1[0] + \mu_1[t]\,\mu_0'[t]}{\mu_0[t]}\right\}$$

The initial values for μ_0 and μ_1:

In[7]:=
```
initsMu = {μ0[0] → 0, μ0'[0] → 2 a[0], μ1'[0] → 0};
```

Verify that the fundamental solution (2.4) - (2.9) satisfies the Riccati-type system:

```
Timing[Together[
  (#1 - #2 &@@@ riccati) /. x : (α | β | γ | δ | ϵ | κ | μ) → x0 //. Join[eqns24to29, eqns210to211]]]
```

```
{10.203, {0, 0, 0, 0, 0, 0, 0}}
```

■ Compute the fundamental solution (2.4) - (2.9) of the Riccati-type system

```
TableForm [eqns = Equal @@@ (Take[riccati, 6] /. x : (α | β | γ | δ | ϵ | κ) → x0)]
```

$$\alpha_0'[t] == -b[t] - 2\,c[t]\,\alpha_0[t] - 4\,a[t]\,\alpha_0[t]^2$$
$$\beta_0'[t] == -c[t]\,\beta_0[t] - 4\,a[t]\,\alpha_0[t]\,\beta_0[t]$$
$$\gamma_0'[t] == -a[t]\,\beta_0[t]^2$$
$$\delta_0'[t] == f[t] + 2\,g[t]\,\alpha_0[t] - c[t]\,\delta_0[t] - 4\,a[t]\,\alpha_0[t]\,\delta_0[t]$$
$$\epsilon_0'[t] == g[t]\,\beta_0[t] - 2\,a[t]\,\beta_0[t]\,\delta_0[t]$$
$$\kappa_0'[t] == g[t]\,\delta_0[t] - a[t]\,\delta_0[t]^2$$

```
(* My simplification procedure for integrals. *)
MySimp [expr_] := FullSimplify[FullSimplify[expr] //. {
        (* Rewrite integrals in terms  of lambda  if possible. *)
        Integrate[a1_. * c[x_] + a2_. * d[x_] + a3_., {x_, a4_, a5_}] /; a2 / a1 == -2 &&
            FreeQ[a1, x] :> a1 * (Log[λ[a4]] - Log[λ[a5]]) + Integrate[a3, {x, a4, a5}],
        (* Extract constant factors. *)
        Integrate[a1_ * a2_, {x_, a3__}] /; FreeQ[a1, x] :> a1 * Integrate[a2, {x, a3}],
        (* Combine  integrals with different ranges. *)
        Integrate[a1_, {x_, a2_, a3_}] + Integrate[a1_, {x_, a3_, a4_}] :>
          Integrate[a1, {x, a2, a4}]} /.
    K[
        _] →
      s]
(* Start with some  substitutions *)
TableForm [
  subs = Join[{#1[t] → #2[t] &@@ eqns24to29[[1]]}, eqns210to211[[{1, 2}]],
      {λ'[t] → (-c[t] + 2 d[t]) λ[t], μ0''[t] → τ[t] * μ0'[t] - 4 * σ[t] * μ0[t]}] /. Pattern → (# &)]
```

$\alpha_0[t] \to -\frac{d[t]}{2 a[t]} + \frac{\mu_0'[t]}{4 a[t] \mu_0[t]}$

$\sigma[t] \to \frac{1}{4}\left(4 a[t] b[t] - 4 c[t] d[t] + 4 d[t]^2 + \frac{2 d[t] a'[t]}{a[t]} - 2 d'[t]\right)$

$\tau[t] \to -2 c[t] + 4 d[t] + \frac{a'[t]}{a[t]}$

$\lambda'[t] \to (-c[t] + 2 d[t]) \lambda[t]$

$\mu_0''[t] \to -4 \sigma[t] \mu_0[t] + \tau[t] \mu_0'[t]$

```
MySimp [DSolve[eqns[[2]] /. subs, β0[t], t]]
```

$\left\{\left\{\beta_0[t] \to \frac{C[1] \lambda[t] \mu_0[1]}{\lambda[1] \mu_0[t]}\right\}\right\}$

```
(* By choosing C[1] appropriately,
we get our solution for beta_0, see Equation (2.5). *)
%[[1, 1]] /. C[1] → -λ[1] / μ0[1]
AppendTo[subs, %];
```

$\beta_0[t] \to -\frac{\lambda[t]}{\mu_0[t]}$

```
(* Now gamma_  0: this doesn't give what we want. *)
MySimp [DSolve[eqns[[3]] /. subs, γ0[t], t]]
```

$\left\{\left\{\gamma_0[t] \to C[1] - \int_1^t \frac{a[s] \lambda[s]^2}{\mu_0[s]^2} ds\right\}\right\}$

```
(* Trick: use the Wronskian. *)
eqns[[3]] /. subs /. a[t] * λ[t]^2 → (μ1[t] μ0'[t] - μ0[t] * μ1'[t]) / (2 μ1[0])
```

$\gamma_0'[t] == -\frac{\mu_1[t] \mu_0'[t] - \mu_0[t] \mu_1'[t]}{2 \mu_0[t]^2 \mu_1[0]}$

```
(* Appropriate choice of C[1] gives the solution gamma_  0, see Equation (2.6). *)
DSolve[%, γ0[t], t] /. C[1] → d[0] / (2 a[0])
```

$\left\{\left\{\gamma_0[t] \to \frac{d[0]}{2 a[0]} + \frac{\mu_1[t]}{2 \mu_0[t] \mu_1[0]}\right\}\right\}$

```
(* Solve the equation for delta_0. *)
MySimp [DSolve[eqns[[4]] /. subs, δ0[t], t]]
```

$\left\{\left\{\delta_0[t] \to \left(\lambda[t] \left(\left(\int_1^t ((2 (a[s] f[s] - d[s] g[s]) \mu_0[s] + g[s] \mu_0'[s]) / (a[s] \lambda[s])) ds\right) \lambda[1] + 2 C[1] \mu_0[1]\right)\right) \Big/ (2 \lambda[1] \mu_0[t])\right\}\right\}$

```
(* Choose C[1]. This is the solution for delta_0 given in (2.7). *)
MySimp [%/.C[1]→λ[1]/(2μ0[1])*
       Integrate[(2 (a[s] f[s] - d[s] g[s]) μ0[s] + g[s] μ0'[s]) / (a[s] λ[s]), {s, 0, 1}]]
```

$$\left\{\left\{\delta_0[t]\to\frac{1}{2\mu_0[t]}\left(\int_0^t((2(a[s]f[s]-d[s]g[s])\mu_0[s]+g[s]\mu_0'[s])/(a[s]\lambda[s]))\,ds\right)\lambda[t]\right\}\right\}$$

```
(* Add the equation for delta to our list of substitutions *)
AppendTo[subs, eqns[[4]] /. Equal→Rule];
```

```
(* Now epsilon_0: this is not what we want (has mu0  in the denominator). *)
MySimp [DSolve[eqns[[5]] /. subs, ϵ0[t], t]]
```

$$\left\{\left\{\epsilon_0[t]\to C[1]-\int_1^t\frac{\lambda[s]\,(g[s]-2a[s]\,\delta_0[s])}{\mu_0[s]}\,ds\right\}\right\}$$

```
(* The solution for epsilon_0. *)
soleps = eqns24to29[[6, 2]][t]
```

$$8\int_0^t\frac{a[s]\,\lambda[s]\,\sigma[s]\,\delta_0[s]\,\mu_0[s]}{\mu_0'[s]^2}\,ds+2\int_0^t\frac{a[s]\left(f[s]-\frac{d[s]\,g[s]}{a[s]}\right)\lambda[s]}{\mu_0'[s]}\,ds-\frac{2a[t]\,\lambda[t]\,\delta_0[t]}{\mu_0'[t]}$$

```
(* Check again that it satisfies the equation. *)
Factor[{eqns[[5, 2]], D[soleps, t]} //. subs]
```

$$\left\{-\frac{\lambda[t]\,(g[t]-2a[t]\,\delta_0[t])}{\mu_0[t]},\ \frac{\lambda[t]\,(-g[t]+2a[t]\,\delta_0[t])}{\mu_0[t]}\right\}$$

```
(* The solution for epsilon'[t] presented in the paper,
but put under a single integral sign. *)
(* This is only the integrand. *)
result = Together[D[eqns24to29[[6, 2]][t], t]]
```

$$\frac{1}{\mu_0'[t]^2}2\,(4a[t]\,\lambda[t]\,\sigma[t]\,\delta_0[t]\,\mu_0[t]+a[t]\,f[t]\,\lambda[t]\,\mu_0'[t]-d[t]\,g[t]\,\lambda[t]\,\mu_0'[t]-\lambda[t]\,\delta_0[t]\,a'[t]\,\mu_0'[t]-a[t]\,\delta_0[t]\,\lambda'[t]\,\mu_0'[t]-a[t]\,\lambda[t]\,\delta_0'[t]\,\mu_0'[t]+a[t]\,\lambda[t]\,\delta_0[t]\,\mu_0''[t])$$

```
(* Test it. *)
Together[eqns[[5, 2]] - result //. subs]
```

0

```
(* The right-hand side of the equation epsilon'[t] = ... *)
eqn5 = eqns[[5, 2]] /. x_[t]→x
```

$$g\beta_0-2a\beta_0\delta_0$$

```
(* The substitutions generate an ideal in some  multivariate polynomial  ring. *)
ideal = Numerator [Together[#1 - #2 &@@@subs /. x_[t]→x]]
```

$$\{2d\mu_0+4a\alpha_0\mu_0-\mu_0',\ -2a^2b+2acd-2ad^2+2a\sigma-da'+ad',\ 2ac-4ad+a\tau-a',\ c\lambda-2d\lambda+\lambda',\ 4\sigma\mu_0-\tau\mu_0'+\mu_0'',\ \lambda+\beta_0\mu_0,\ -f-2g\alpha_0+c\delta_0+4a\alpha_0\delta_0+\delta_0'\}$$

```
(* We can prove correctness of the given solution by reduction with Groebner basis. *)
vars = #[[1, 0]] & /@ subs;
Last[PolynomialReduce [
   Together[(μ0') ^2 * (eqn5 - result /. x_[t]→x)], GroebnerBasis[ideal, vars], vars]]
```

0

```
(* Similarly, we can find a different, but equivalent, expression for epsilon'[t]. *)
(* Play around with vars to influence which variables appear in the result. *)
vars = {δ0', β0, b, c};
Factor[Last[PolynomialReduce [μ0' * eqn5, GroebnerBasis[ideal, vars], vars]] / μ0']
```

$$-\frac{2\lambda\,(d+2a\alpha_0)\,(g-2a\delta_0)}{\mu_0'}$$

```
(* Together with the desired initialcondition, we get this solution for epsilon_0. *)
ϵ0→Function[t, -g[0] / (2 a[0]) -
      2 Integrate[λ[s] / μ0'[s] * (d[s] + 2 a[s] α0[s]) * (g[s] - 2 a[s] δ0[s]), {s, 0, t}]]
```

$$\epsilon_0 \to \text{Function}\left[t, -\frac{g[0]}{2\,a[0]} - 2\int_0^t \frac{1}{\mu_0'[s]} \lambda[s]\,(d[s] + 2\,a[s]\,\alpha_0[s])\,(g[s] - 2\,a[s]\,\delta_0[s])\,ds\right]$$

```
(* Test it. *)
Together[(#1 - #2 &@@ eqns[[5]]) /. % //. subs]
```

0

```
(* Another different, still equivalent, expression for epsilon_0'[t], free of alpha_0. *)
vars = {α0, β0, δ0', b, c};
Factor[Last[PolynomialReduce [μ0' * eqn5, GroebnerBasis[ideal, vars], vars]] / μ0']
```

$$\frac{2\,(a\,f\,\lambda - d\,g\,\lambda + a\,\delta_0\,\lambda' - a\,\lambda\,\delta_0')}{\mu_0'}$$

```
(* In order to get rid of delta', we can do integration by parts. *)
Factor[% - (-2 a λ δ0' / μ0') - δ0 * (-2 (a' λ + a λ') μ0' - (-2 a λ) μ0'') / (μ0') ^2]
```

$$\frac{1}{(\mu_0')^2} 2\,(a\,f\,\lambda\,\mu_0' - d\,g\,\lambda\,\mu_0' + \lambda\,\delta_0\,a'\,\mu_0' + 2\,a\,\delta_0\,\lambda'\,\mu_0' - a\,\lambda\,\delta_0\,\mu_0'')$$

```
(* Perform  some  of our substitutions and get the integrand of Equation (2.9). *)
Factor[% //. (subs /. x_[t] → x) /. First[Solve[(#1 - #2 &@@ subs[[2]] /. x_[t] → x) == 0, d']]]
```

$$\frac{2\,\lambda\,(4\,a\,\sigma\,\delta_0\,\mu_0 + a\,f\,\mu_0' - d\,g\,\mu_0')}{(\mu_0')^2}$$

```
(* Together with the desired initialcondition, we get this solution for epsilon_0. *)
ϵ0→Function[t, -g[0] / (2 a[0]) + 2 Integrate[
        (λ[s] * (a[s] * (f[s] - δ0'[s]) - d[s] * g[s]) + a[s] * δ0[s] * λ'[s]) / μ0'[s], {s, 0, t}]]
```

$$\epsilon_0 \to \text{Function}\left[t, -\frac{g[0]}{2\,a[0]} + 2\int_0^t \frac{1}{\mu_0'[s]} (\lambda[s]\,(a[s]\,(f[s] - \delta_0'[s]) - d[s]\,g[s]) + a[s]\,\delta_0[s]\,\lambda'[s])\,ds\right]$$

```
(* Test it. *)
Together[(#1 - #2 &@@ eqns[[5]]) /. % //. subs]
```

0

```
(* Finally, consider kappa_0. *)
MySimp [DSolve[eqns[[6]] /. subs, κ0[t], t]]
```

$$\left\{\left\{\kappa_0[t] \to C[1] + \int_1^t \delta_0[s]\,(g[s] - a[s]\,\delta_0[s])\,ds\right\}\right\}$$

```
(* C[1] is chosen such that we get 0 as lower integration bound. *)
MySimp [% /. C[1] → Integrate[δ0[s] (g[s] - a[s] δ0[s]), {s, 0, 1}]]
```

$$\left\{\left\{\kappa_0[t] \to \int_0^t \delta_0[s]\,(g[s] - a[s]\,\delta_0[s])\,ds\right\}\right\}$$

■ Derive the formula (2.12) for the Wronskian

```
(* Define the Wronskian and make  an ansatz for its differential equation. *)
W[t_] := Module[{z}, Det[Table[D[μi[z], {z, j}], {i, 0, 1}, {j, 0, 1}]] /. z → t];
ansatz = c0 * w[t] + c1 * D[w[t], t];
ansatz /. w → W /. Derivative[2][μ_][t] → τ[t] * D[μ[t], t] - 4 * σ[t] * μ[t]
```

$$c0\,(-\mu_1[t]\,\mu_0'[t] + \mu_0[t]\,\mu_1'[t]) +$$
$$c1\,(-\mu_1[t]\,(-4\,\sigma[t]\,\mu_0[t] + \tau[t]\,\mu_0'[t]) + \mu_0[t]\,(-4\,\sigma[t]\,\mu_1[t] + \tau[t]\,\mu_1'[t]))$$

```
(* Conditions on the unknown coefficients. *)
DeleteCases[Flatten[CoefficientList[%, Variables[W[t]]]], 0]
```

$$\{c0 + c1\,\tau[t],\ -c0 - c1\,\tau[t]\}$$

```
(* This is the ODE for the Wronskian. *)
deW = ansatz /. c0 → -c1 * τ[t] /. c1 → 1 /. τ[t] → D[a[t], t] / a[t] - 2 * c[t] + 4 * d[t]
```

$$-w[t]\left(-2\,c[t] + 4\,d[t] + \frac{a'[t]}{a[t]}\right) + w'[t]$$

```
(* Its solution. *)
wronsk = ExpandAll[DSolve[deW == 0, w[t], t][[1, 1, 2]] /. K[1] → z]
```

$$e^{\int_1^t\left(-2\,c[z]+4\,d[z]+\frac{a'[z]}{a[z]}\right)dz}\,C[1]$$

```
(* Expand the integral. *)wronsk = wronsk //. HoldPattern[Integrate[a_ + b_, {z, a1_, a2_}]] :→
        Integrate[a, {z, a1, a2}] + Integrate[b, {z, a1, a2}] /.
    HoldPattern[Integrate[a1_. * a2_, {z, 1, t}] /; FreeQ[a1, z]] :→
      a1 * (Integrate[a2, {z, 0, t}] - Integrate[a2, {z, 0, 1}])
```

$$\frac{1}{a[1]}\,e^{-2\left(-\int_0^1 c[z]\,dz+\int_0^t c[z]\,dz\right)+4\left(-\int_0^1 d[z]\,dz+\int_0^t d[z]\,dz\right)}\,a[t]\,C[1]$$

```
(* Use definition of lambda and simplify. *)
wronsk = ExpandAll[wronsk /.
      HoldPattern[Integrate[c[z], {z, 0, t}]] :→ 2 * Integrate[d[z], {z, 0, t}] - Log[λ[t]]]
```

$$\frac{e^{2\int_0^1 c[z]\,dz-4\int_0^1 d[z]\,dz}\,a[t]\,C[1]\,\lambda[t]^2}{a[1]}$$

```
(* Delete all factors free of t and subsume them into the new constant C[1]. *)
wronsk = C[1] * DeleteCases[wronsk, a_ /; FreeQ[a, t]]
```

$$a[t]\,C[1]\,\lambda[t]^2$$

```
Solve[(W[t] - wronsk == 0) /. t → 0 /. initsMu /. λ[0] → 1, C[1]]
```

$$\{\{C[1] \to -2\,\mu_1[0]\}\}$$

```
(* This is the right-hand side of (2.12), used above. *)
wronsk /. First[%]
```

$$-2\,a[t]\,\lambda[t]^2\,\mu_1[0]$$

■ Asymptotic Expansions (2.13)

Now compute the asymptotics of the fundamental solution. The result corresponds exactly to Equation (2.13):

```
(* This takes some  time  since Mathematica  tries to simplify the integrals; *)
(* the trick with Hold speeds it up. *)
Timing[
  exprs = ReleaseHold[Hold@@{Last[#][t] & /@eqns24to29} //. eqns24to29 /. initsMu /.
          (* Substitute the Taylor expansion of mu . *)
          (μ : μ0 | μ1) [t_] :→ μ[0] + μ'[0] * t + μ''[0] / 2 * t^2 /. initsMu];
  (* Now the series expansions of the above expressions. *)
  ser = Series[exprs, {t, 0, 0}] //. eqns210to211 /. initsMu;
]
TableForm [MapThread[Rule, {First[#][t] & /@eqns24to29, ExpandAll[ser]}]]
```

{71.574, Null}

$\alpha_0[t] \to \frac{1}{4\,a[0]\,t} + \left(-\frac{c[0]}{4\,a[0]} - \frac{a'[0]}{8\,a[0]^2}\right) + O[t]^1$

$\beta_0[t] \to -\frac{1}{2\,a[0]\,t} + \frac{a'[0]}{4\,a[0]^2} + O[t]^1$

$\lambda[t] \to 1 + O[t]^1$

$\gamma_0[t] \to \frac{1}{4\,a[0]\,t} + \left(\frac{c[0]}{4\,a[0]} - \frac{a'[0]}{8\,a[0]^2}\right) + O[t]^1$

$\delta_0[t] \to \frac{g[0]}{2\,a[0]} + O[t]^1$

$\epsilon_0[t] \to -\frac{g[0]}{2\,a[0]} + O[t]^1$

$\kappa_0[t] \to O[t]^1$

2.1.2 Special Beam Modes in Weakly Inhomogeneous Media

■ Generalized Hermite-Gaussian beams and Ermakov-type system

In[8]:=
```
(* This is equation (2.15). *)
HG[x_, t_] :=
    Exp[I * (α[t] * x^2 + δ[t] * x + κ[t] + (2 * n + 1) * γ[t])] / Sqrt[μ[t] * 2^n * n! * Sqrt[Pi]] *
      Exp[-(β[t] * x + ϵ[t])^2 / 2] * HermiteH [n, β[t] * x + ϵ[t]];
TraditionalForm[HG[x, t] /. x_[t] → x]
```

Out[9]//TraditionalForm=

$$\left(H_n(\beta\,x + \epsilon)\exp\left(-\frac{1}{2}(\beta\,x + \epsilon)^2 + i\left(\kappa + \gamma\,(2\,n + 1) + \alpha\,x^2 + \delta\,x\right)\right)\right) \Big/ \left(\sqrt[4]{\pi}\,\sqrt{\mu\,2^n\,n!}\right)$$

In[10]:=
```
(* These are equations (2.16) - (2.22). *)
eqns216to222 = {
    μ → Function[t, μ[0] * μ0[t] * Sqrt[β[0]^4 + 4 * (α[0] + γ0[t])^2]],
    α → Function[t, α0[t] - β0[t]^2 * ((α[0] + γ0[t]) / (β[0]^4 + 4 * (α[0] + γ0[t])^2))],
    β → Function[t, -β[0] * β0[t] / Sqrt[β[0]^4 + 4 * (α[0] + γ0[t])^2]],
    γ → Function[t, γ[0] - 1/2 * ArcTan[β[0]^2 / (2 * (α[0] + γ0[t]))]],
    δ → Function[t, δ0[t] - β0[t] * (ϵ[0] * β[0]^3 + 2 * (α[0] + γ0[t]) * (δ[0] + ϵ0[t])) /
           (β[0]^4 + 4 * (α[0] + γ0[t])^2)],
    ϵ → Function[t, (2 * ϵ[0] * (α[0] + γ0[t]) - β[0] * (δ[0] + ϵ0[t])) /
        Sqrt[β[0]^4 + 4 * (α[0] + γ0[t])^2]],
    κ → Function[t, κ[0] + κ0[t] - ϵ[0] * β[0]^3 * (δ[0] + ϵ0[t]) / (β[0]^4 + 4 * (α[0] + γ0[t])^2) +
        (α[0] + γ0[t]) *
         ((ϵ[0]^2 * β[0]^2 - (δ[0] + ϵ0[t])^2) / (β[0]^4 + 4 * (α[0] + γ0[t])^2))]};
TableForm [(#1[t] == #2[t] &@@@ eqns216to222) /. x_[t] → x] // TraditionalForm
```

Out[11]//TraditionalForm=

$$\mu = \mu_0\, \mu(0) \sqrt{4\,(\alpha(0) + \gamma_0)^2 + \beta(0)^4}$$
$$\alpha = \alpha_0 - \frac{\beta_0^2\,(\alpha(0)+\gamma_0)}{4\,(\alpha(0)+\gamma_0)^2+\beta(0)^4}$$
$$\beta = -\frac{\beta_0\,\beta(0)}{\sqrt{4\,(\alpha(0)+\gamma_0)^2+\beta(0)^4}}$$
$$\gamma = \gamma(0) - \frac{1}{2}\tan^{-1}\!\left(\frac{\beta(0)^2}{2\,(\alpha(0)+\gamma_0)}\right)$$
$$\delta = \delta_0 - \frac{\beta_0\left(2\,(\alpha(0)+\gamma_0)\,(\delta(0)+\epsilon_0)+\beta(0)^3\,\epsilon(0)\right)}{4\,(\alpha(0)+\gamma_0)^2+\beta(0)^4}$$
$$\epsilon = \frac{2\,\epsilon(0)\,(\alpha(0)+\gamma_0)-\beta(0)\,(\delta(0)+\epsilon_0)}{\sqrt{4\,(\alpha(0)+\gamma_0)^2+\beta(0)^4}}$$
$$\kappa = -\frac{\beta(0)^3\,\epsilon(0)\,(\delta(0)+\epsilon_0)}{4\,(\alpha(0)+\gamma_0)^2+\beta(0)^4} + \frac{(\alpha(0)+\gamma_0)\left(\beta(0)^2\,\epsilon(0)^2-(\delta(0)+\epsilon_0)^2\right)}{4\,(\alpha(0)+\gamma_0)^2+\beta(0)^4} + \kappa_0 + \kappa(0)$$

First we verify that (2.15) - (2.22) satisfies the Schrödinger equation (2.1), using the fact that α_0, ..., κ_0 are the fundamental solution of the Riccati-type system.

```
Timing[Together[
    (* Plug (2.15) into (2.1). Use equations (2.16) -
      (2.22) and the Riccati-type system . *)
    schroedinger/. ψ → HG /. eqns216to222 /. (riccati/. x : (α | β | γ | δ | ϵ | κ | μ) → x0) /.
      (* Rewrite H_{n-2} in terms  of H_{n-1} and H_n. *)
      HermiteH [n - 2, x_] :→ (2 x * HermiteH [n - 1, x] - HermiteH [n, x]) / (2 n - 2)
  ]]
```

```
{1.653, 0}
```

In[12]:=
```
(* This is the Ermakov -type system , basically Equations (2.41) - (2.47), see below. *)
ermakov = MapThread[#1[[1]] → #1[[2]] + #2 &,
      {riccati, c0 * a[t] * β[t]^2 * {β[t]^2, 0, 0, 2 β[t] * ϵ[t], 0, ϵ[t]^2, 0}}];
TableForm [ermakov /. x_[t] → x]
```

Out[13]//TableForm=

```
α′ → -b - 2 c α - 4 a α^2 + a β^4 c0
β′ → -c β - 4 a α β
γ′ → -a β^2
δ′ → f + 2 g α - c δ - 4 a α δ + 2 a β^3 ϵ c0
ϵ′ → g β - 2 a β δ
κ′ → g δ - a δ^2 + a β^2 ϵ^2 c0
μ′ → 2 (d μ + 2 a α μ)
```

Now verify that the expressions (2.16) - (2.22) satisfy the Ermakov-type system with $c_0 = 1$. Use the fact that α_0, ..., κ_0 are the fundamental solution of the Riccati-type system with $c_0 = 0$.

```
Together[(#1 - #2 &@@@ Take[ermakov , 6]) /. c0 → 1 /. eqns216to222 /.
    (riccati/. x : (α | β | γ | δ | ϵ | κ | μ) → x0)]
```

```
{0, 0, 0, 0, 0, 0}
```

■ Generalized Airy beams and Riccati-type system

In[14]:=
```
(* This is equation (2.23). *)
AB[x_, t_] :=
  Exp[I * (x^2 * α[t] + x * δ[t] + κ[t] - γ[t] * (x * β[t] + ϵ[t] - 2 / 3 * γ[t]^2))] / Sqrt[μ[t]] *
   AiryAi[x * β[t] + ϵ[t] - γ[t]^2];
TraditionalForm[AB[x, t] /. x_[t] → x]
```

Out[15]//TraditionalForm=

$$\frac{1}{\sqrt{\mu}}\text{Ai}\left(-\gamma^2 + x\beta + \epsilon\right)\exp\left(i\left(\kappa + \alpha x^2 - \gamma\left(-\frac{2\gamma^2}{3} + \beta x + \epsilon\right) + \delta x\right)\right)$$

In[16]:=
```
(* These are equations (2.24) - (2.30). *)
eqns224to230 = {
    μ → Function[t, 2 * μ[0] * μ0[t] * (α[0] + γ0[t])],
    α → Function[t, α0[t] - (β0[t])^2 / (4 * (α[0] + γ0[t]))],
    β → Function[t, -(β[0] * β0[t]) / (2 * (α[0] + γ0[t]))],
    γ → Function[t, γ[0] - (β[0]^2) / (4 * (α[0] + γ0[t]))],
    δ → Function[t, δ0[t] - (β0[t] * (δ[0] + ϵ0[t])) / (2 * (α[0] + γ0[t]))],
    ϵ → Function[t, ϵ[0] - (β[0] * (δ[0] + ϵ0[t])) / (2 * (α[0] + γ0[t]))],
    κ → Function[t, κ[0] + κ0[t] - ((δ[0] + ϵ0[t])^2) / (4 * (α[0] + γ0[t]))]};
TableForm[(#1[t] == #2[t] &@@@ eqns224to230)] // TraditionalForm
```

Out[17]//TraditionalForm=

$$\mu(t) = 2\,\mu(0)\,\mu_0(t)\,(\alpha(0) + \gamma_0(t))$$
$$\alpha(t) = \alpha_0(t) - \frac{\beta_0(t)^2}{4\,(\alpha(0)+\gamma_0(t))}$$
$$\beta(t) = -\frac{\beta(0)\,\beta_0(t)}{2\,(\alpha(0)+\gamma_0(t))}$$
$$\gamma(t) = \gamma(0) - \frac{\beta(0)^2}{4\,(\alpha(0)+\gamma_0(t))}$$
$$\delta(t) = \delta_0(t) - \frac{\beta_0(t)\,(\delta(0)+\epsilon_0(t))}{2\,(\alpha(0)+\gamma_0(t))}$$
$$\epsilon(t) = \epsilon(0) - \frac{\beta(0)\,(\delta(0)+\epsilon_0(t))}{2\,(\alpha(0)+\gamma_0(t))}$$
$$\kappa(t) = \kappa(0) - \frac{(\delta(0)+\epsilon_0(t))^2}{4\,(\alpha(0)+\gamma_0(t))} + \kappa_0(t)$$

Verify that Equations (2.23) - (2.30) are a solution of the Schrödinger equation (2.1).

```
Timing[Together[
  schroedinger /. ψ → AB /. eqns224to230 /. (riccati /. x : (α | β | γ | δ | ϵ | κ | μ) → x0)
 ]]
```

```
{0.639, 0}
```

Plug the solutions (2.24) - (2.30) into the Riccati-type system, i.e. $c_0 = 0$, and use the fact that α_0, ..., κ_0 are the fundamental solution of the Riccati-type system.

```
Together[(#1 - #2 &@@@ Take[riccati, 6]) /. eqns224to230 /. (riccati /. x : (α | β | γ | δ | ϵ | κ | μ) → x0)]
```

```
{0, 0, 0, 0, 0, 0}
```

2.2.2 Cylindrical Symmetry, Proof of Lemma 1

In[18]:=
```
(* The nonlinear parabolic equation (2.39) in 2D. *)
eqn239 = -I * D[A[x, y, s], s] - a[s] * (D[A[x, y, s], x, x] + D[A[x, y, s], y, y]) +
    b[s] * (x^2 + y^2) * A[x, y, s] - I * c[s] * (x * D[A[x, y, s], x] + y * D[A[x, y, s], y]) -
    2 I * d[s] * A[x, y, s] - (x * f1[s] + y * f2[s]) * A[x, y, s] +
    I * (g1[s] * D[A[x, y, s], x] + g2[s] * D[A[x, y, s], y]) + h[s] * Abs[A[x, y, s]]^p * A[x, y, s];
```

```
(* Substitute (2.38) into Equation (2.39). *)
expr = Together[(eqn239 /.
      A → Function[{x, y, s}, Exp[I * (α[s] * (x^2 + y^2) + δ1[s] * x + δ2[s] * y + κ1[s] + κ2[s])] *
          χ[β[s] * x + ϵ1[s], β[s] * y + ϵ2[s], γ[s]] / μ[s]] /.
     h[s] → h0 * a[s] * β[s]^2 * μ[s]^p) / Exp[
    I * (α[s] * (x^2 + y^2) + δ1[s] * x + δ2[s] * y + κ1[s] + κ2[s])]]
```

$$
\begin{aligned}
&\frac{1}{\mu[s]^2}\Big(x^2 b[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]+\\
&\quad y^2 b[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad 2\,\mathbb{i}\,d[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad 4\,\mathbb{i}\,a[s]\,\alpha[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]+\\
&\quad 2x^2 c[s]\,\alpha[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]+\\
&\quad 2y^2 c[s]\,\alpha[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]+\\
&\quad 4x^2 a[s]\,\alpha[s]^2\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]+\\
&\quad 4y^2 a[s]\,\alpha[s]^2\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]+\\
&\quad \left(e^{-\mathrm{Im}\;[(x^2+y^2)\,\alpha[s]+x\delta_1[s]+y\delta_2[s]+\kappa_1[s]+\kappa_2[s]]}\right)^p a[s]\,\mathrm{Abs}\Big[\frac{1}{\mu[s]}\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\Big]^p\\
&\qquad h_0\,\beta[s]^2\mu[s]^{1+p}\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad x\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,f_1[s]-\\
&\quad y\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,f_2[s]-\\
&\quad 2x\alpha[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,g_1[s]-\\
&\quad 2y\alpha[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,g_2[s]+\\
&\quad x c[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\delta_1[s]+\\
&\quad 4x a[s]\,\alpha[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\delta_1[s]-\\
&\quad \mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,g_1[s]\,\delta_1[s]+\\
&\quad a[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\delta_1[s]^2+\\
&\quad y c[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\delta_2[s]+\\
&\quad 4y a[s]\,\alpha[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\delta_2[s]-\\
&\quad \mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,g_2[s]\,\delta_2[s]+\\
&\quad a[s]\,\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\delta_2[s]^2+\\
&\quad x^2\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\alpha'[s]+\\
&\quad y^2\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\alpha'[s]+\\
&\quad \mathbb{i}\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\mu'[s]+\\
&\quad x\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\delta_1'[s]+\\
&\quad y\mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\delta_2'[s]+\\
&\quad \mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\kappa_1'[s]+\\
&\quad \mu[s]\,\chi[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\,\kappa_2'[s]-\\
&\quad \mathbb{i}\,\mu[s]\,\gamma'[s]\,\chi^{(0,0,1)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad \mathbb{i}\,y c[s]\,\beta[s]\,\mu[s]\,\chi^{(0,1,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad 4\,\mathbb{i}\,y a[s]\,\alpha[s]\,\beta[s]\,\mu[s]\,\chi^{(0,1,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]+\\
&\quad \mathbb{i}\,\beta[s]\,\mu[s]\,g_2[s]\,\chi^{(0,1,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad 2\,\mathbb{i}\,a[s]\,\beta[s]\,\mu[s]\,\delta_2[s]\,\chi^{(0,1,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad \mathbb{i}\,y\mu[s]\,\beta'[s]\,\chi^{(0,1,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad \mathbb{i}\,\mu[s]\,\epsilon_2'[s]\,\chi^{(0,1,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad a[s]\,\beta[s]^2\mu[s]\,\chi^{(0,2,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad \mathbb{i}\,x c[s]\,\beta[s]\,\mu[s]\,\chi^{(1,0,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad 4\,\mathbb{i}\,x a[s]\,\alpha[s]\,\beta[s]\,\mu[s]\,\chi^{(1,0,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]+\\
&\quad \mathbb{i}\,\beta[s]\,\mu[s]\,g_1[s]\,\chi^{(1,0,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad 2\,\mathbb{i}\,a[s]\,\beta[s]\,\mu[s]\,\delta_1[s]\,\chi^{(1,0,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad \mathbb{i}\,x\mu[s]\,\beta'[s]\,\chi^{(1,0,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad \mathbb{i}\,\mu[s]\,\epsilon_1'[s]\,\chi^{(1,0,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]-\\
&\quad a[s]\,\beta[s]^2\mu[s]\,\chi^{(2,0,0)}[x\beta[s]+\epsilon_1[s],\,y\beta[s]+\epsilon_2[s],\,\gamma[s]]\Big)
\end{aligned}
$$

```
(* Short-hand notation: omit the arguments of χ and the dependencies on s. *)
expr = expr /. χ[__] → χ /. Derivative[a__][χ][__] → Derivative[a][χ] /. a1_[s] → a1
```

$$\frac{1}{\mu^2}\Big(-2 i d\mu\chi + b x^2\mu\chi + b y^2\mu\chi - 4 i a\alpha\mu\chi + 2 c x^2\alpha\mu\chi + 2 c y^2\alpha\mu\chi + 4 a x^2\alpha^2\mu\chi + 4 a y^2\alpha^2\mu\chi - x\mu\chi f_1 - y\mu\chi f_2 - 2 x\alpha\mu\chi g_1 - 2 y\alpha\mu\chi g_2 + a\left(e^{-\mathrm{Im}\,[(x^2+y^2)\alpha + x\delta_1 + y\delta_2 + \kappa_1 + \kappa_2]}\right)^p \beta^2\mu^{1+p}\chi\,\mathrm{Abs}\left[\frac{\chi}{\mu}\right]^p h_0 + c x\mu\chi\delta_1 + 4 a x\alpha\mu\chi\delta_1 - \mu\chi g_1\delta_1 + a\mu\chi\delta_1^2 + c y\mu\chi\delta_2 + 4 a y\alpha\mu\chi\delta_2 - \mu\chi g_2\delta_2 + a\mu\chi\delta_2^2 + x^2\mu\chi\alpha' + y^2\mu\chi\alpha' + i\chi\mu' + x\mu\chi\delta_1' + y\mu\chi\delta_2' + \mu\chi\kappa_1' + \mu\chi\kappa_2' - i\mu\gamma'\chi^{(0,0,1)} - i c y\beta\mu\chi^{(0,1,0)} - 4 i a y\alpha\beta\mu\chi^{(0,1,0)} + i\beta\mu g_2\chi^{(0,1,0)} - 2 i a\beta\mu\delta_2\chi^{(0,1,0)} - i y\mu\beta'\chi^{(0,1,0)} - i\mu\epsilon_2'\chi^{(0,1,0)} - a\beta^2\mu\chi^{(0,2,0)} - i c x\beta\mu\chi^{(1,0,0)} - 4 i a x\alpha\beta\mu\chi^{(1,0,0)} + i\beta\mu g_1\chi^{(1,0,0)} - 2 i a\beta\mu\delta_1\chi^{(1,0,0)} - i x\mu\beta'\chi^{(1,0,0)} - i\mu\epsilon_1'\chi^{(1,0,0)} - a\beta^2\mu\chi^{(2,0,0)}\Big)$$

```
(* Since all functions in question are real-valued,
we can simplify the Im [...] part: *)
expr = expr /. Im [_] → 0
```

$$\frac{1}{\mu^2}\Big(-2 i d\mu\chi + b x^2\mu\chi + b y^2\mu\chi - 4 i a\alpha\mu\chi + 2 c x^2\alpha\mu\chi + 2 c y^2\alpha\mu\chi + 4 a x^2\alpha^2\mu\chi + 4 a y^2\alpha^2\mu\chi - x\mu\chi f_1 - y\mu\chi f_2 - 2 x\alpha\mu\chi g_1 - 2 y\alpha\mu\chi g_2 + a\beta^2\mu^{1+p}\chi\,\mathrm{Abs}\left[\frac{\chi}{\mu}\right]^p h_0 + c x\mu\chi\delta_1 + 4 a x\alpha\mu\chi\delta_1 - \mu\chi g_1\delta_1 + a\mu\chi\delta_1^2 + c y\mu\chi\delta_2 + 4 a y\alpha\mu\chi\delta_2 - \mu\chi g_2\delta_2 + a\mu\chi\delta_2^2 + x^2\mu\chi\alpha' + y^2\mu\chi\alpha' + i\chi\mu' + x\mu\chi\delta_1' + y\mu\chi\delta_2' + \mu\chi\kappa_1' + \mu\chi\kappa_2' - i\mu\gamma'\chi^{(0,0,1)} - i c y\beta\mu\chi^{(0,1,0)} - 4 i a y\alpha\beta\mu\chi^{(0,1,0)} + i\beta\mu g_2\chi^{(0,1,0)} - 2 i a\beta\mu\delta_2\chi^{(0,1,0)} - i y\mu\beta'\chi^{(0,1,0)} - i\mu\epsilon_2'\chi^{(0,1,0)} - a\beta^2\mu\chi^{(0,2,0)} - i c x\beta\mu\chi^{(1,0,0)} - 4 i a x\alpha\beta\mu\chi^{(1,0,0)} + i\beta\mu g_1\chi^{(1,0,0)} - 2 i a\beta\mu\delta_1\chi^{(1,0,0)} - i x\mu\beta'\chi^{(1,0,0)} - i\mu\epsilon_1'\chi^{(1,0,0)} - a\beta^2\mu\chi^{(2,0,0)}\Big)$$

```
(* From (2.47) we may conclude that μ[s] is always positive *)
expr = PowerExpand[Numerator [Together[expr /. Abs[χ / μ] → Abs[χ] / μ]]]
```

$$-2 i d\mu\chi + b x^2\mu\chi + b y^2\mu\chi - 4 i a\alpha\mu\chi + 2 c x^2\alpha\mu\chi + 2 c y^2\alpha\mu\chi + 4 a x^2\alpha^2\mu\chi + 4 a y^2\alpha^2\mu\chi - x\mu\chi f_1 - y\mu\chi f_2 - 2 x\alpha\mu\chi g_1 - 2 y\alpha\mu\chi g_2 + a\beta^2\mu\chi\,\mathrm{Abs}[\chi]^p h_0 + c x\mu\chi\delta_1 + 4 a x\alpha\mu\chi\delta_1 - \mu\chi g_1\delta_1 + a\mu\chi\delta_1^2 + c y\mu\chi\delta_2 + 4 a y\alpha\mu\chi\delta_2 - \mu\chi g_2\delta_2 + a\mu\chi\delta_2^2 + x^2\mu\chi\alpha' + y^2\mu\chi\alpha' + i\chi\mu' + x\mu\chi\delta_1' + y\mu\chi\delta_2' + \mu\chi\kappa_1' + \mu\chi\kappa_2' - i\mu\gamma'\chi^{(0,0,1)} - i c y\beta\mu\chi^{(0,1,0)} - 4 i a y\alpha\beta\mu\chi^{(0,1,0)} + i\beta\mu g_2\chi^{(0,1,0)} - 2 i a\beta\mu\delta_2\chi^{(0,1,0)} - i y\mu\beta'\chi^{(0,1,0)} - i\mu\epsilon_2'\chi^{(0,1,0)} - a\beta^2\mu\chi^{(0,2,0)} - i c x\beta\mu\chi^{(1,0,0)} - 4 i a x\alpha\beta\mu\chi^{(1,0,0)} + i\beta\mu g_1\chi^{(1,0,0)} - 2 i a\beta\mu\delta_1\chi^{(1,0,0)} - i x\mu\beta'\chi^{(1,0,0)} - i\mu\epsilon_1'\chi^{(1,0,0)} - a\beta^2\mu\chi^{(2,0,0)}$$

In[19]:=
```
(* Now Equation (2.40), this is the one we want to end up with. *)
eqn240 = -I * Derivative[0, 0, 1][χ] + Derivative[2, 0, 0][χ] +
   Derivative[0, 2, 0][χ] - c0 * ((β * x + ϵ1) ^ 2 + (β * y + ϵ2) ^ 2) * χ - h0 * Abs[χ] ^ p * χ
```

Out[19]= $-\chi\,\mathrm{Abs}[\chi]^p h_0 - \chi c_0\left((x\beta+\epsilon_1)^2 + (y\beta+\epsilon_2)^2\right) - i\chi^{(0,0,1)} + \chi^{(0,2,0)} + \chi^{(2,0,0)}$

```
(* Equate our expression with (2.40) such that |χ|^p disappears. *)
cond = Numerator [Together[eqn240 + expr / Coefficient[expr, h0 * χ * Abs[χ] ^ p]]]
```

$$-2 i d\mu\chi + b x^2\mu\chi + b y^2\mu\chi - 4 i a\alpha\mu\chi + 2 c x^2\alpha\mu\chi + 2 c y^2\alpha\mu\chi + 4 a x^2\alpha^2\mu\chi + 4 a y^2\alpha^2\mu\chi - a x^2\beta^4\mu\chi c_0 - a y^2\beta^4\mu\chi c_0 - x\mu\chi f_1 - y\mu\chi f_2 - 2 x\alpha\mu\chi g_1 - 2 y\alpha\mu\chi g_2 + c x\mu\chi\delta_1 + 4 a x\alpha\mu\chi\delta_1 - \mu\chi g_1\delta_1 + a\mu\chi\delta_1^2 + c y\mu\chi\delta_2 + 4 a y\alpha\mu\chi\delta_2 - \mu\chi g_2\delta_2 + a\mu\chi\delta_2^2 - 2 a x\beta^3\mu\chi c_0\epsilon_1 - a\beta^2\mu\chi c_0\epsilon_1^2 - 2 a y\beta^3\mu\chi c_0\epsilon_2 - a\beta^2\mu\chi c_0\epsilon_2^2 + x^2\mu\chi\alpha' + y^2\mu\chi\alpha' + i\chi\mu' + x\mu\chi\delta_1' + y\mu\chi\delta_2' + \mu\chi\kappa_1' + \mu\chi\kappa_2' - i a\beta^2\mu\chi^{(0,0,1)} - i\mu\gamma'\chi^{(0,0,1)} - i c y\beta\mu\chi^{(0,1,0)} - 4 i a y\alpha\beta\mu\chi^{(0,1,0)} + i\beta\mu g_2\chi^{(0,1,0)} - 2 i a\beta\mu\delta_2\chi^{(0,1,0)} - i y\mu\beta'\chi^{(0,1,0)} - i\mu\epsilon_2'\chi^{(0,1,0)} - i c x\beta\mu\chi^{(1,0,0)} - 4 i a x\alpha\beta\mu\chi^{(1,0,0)} + i\beta\mu g_1\chi^{(1,0,0)} - 2 i a\beta\mu\delta_1\chi^{(1,0,0)} - i x\mu\beta'\chi^{(1,0,0)} - i\mu\epsilon_1'\chi^{(1,0,0)}$$

```
(* Coefficient comparison w.r.t. the partial derivatives of χ, plus w.r.t. x and y. *)
cond = Union[DeleteCases[
    Flatten[CoefficientList[cond /. Derivative[a1_, a2_, a3_][χ] → d1 ^ a1 * d2 ^ a2 * d3 ^ a3 /.
        χ → 1, {d1, d2, d3, x, y}]], 0]]
```

$$\{b\mu + 2 c\alpha\mu + 4 a\alpha^2\mu - a\beta^4\mu c_0 + \mu\alpha',\ -i c\beta\mu - 4 i a\alpha\beta\mu - i\mu\beta',\ -i a\beta^2\mu - i\mu\gamma',\ -\mu f_1 - 2\alpha\mu g_1 + c\mu\delta_1 + 4 a\alpha\mu\delta_1 - 2 a\beta^3\mu c_0\epsilon_1 + \mu\delta_1',\ -\mu f_2 - 2\alpha\mu g_2 + c\mu\delta_2 + 4 a\alpha\mu\delta_2 - 2 a\beta^3\mu c_0\epsilon_2 + \mu\delta_2',\ i\beta\mu g_1 - 2 i a\beta\mu\delta_1 - i\mu\epsilon_1',\ i\beta\mu g_2 - 2 i a\beta\mu\delta_2 - i\mu\epsilon_2',\ -2 i d\mu - 4 i a\alpha\mu - \mu g_1\delta_1 + a\mu\delta_1^2 - \mu g_2\delta_2 + a\mu\delta_2^2 - a\beta^2\mu c_0\epsilon_1^2 - a\beta^2\mu c_0\epsilon_2^2 + i\mu' + \mu\kappa_1' + \mu\kappa_2'\}$$

```
(* Separate real and imaginary parts (all occurring quantities are real). *)
cond = DeleteCases[Flatten[CoefficientList[cond /. Complex [a1_, a2_] :> a1 + i * a2, i]], 0]
```

$\{b\mu + 2c\alpha\mu + 4a\alpha^2\mu - a\beta^4\mu c_0 + \mu\alpha', -c\beta\mu - 4a\alpha\beta\mu - \mu\beta',$
$-a\beta^2\mu - \mu\gamma', -\mu f_1 - 2\alpha\mu g_1 + c\mu\delta_1 + 4a\alpha\mu\delta_1 - 2a\beta^3\mu c_0\epsilon_1 + \mu\delta_1',$
$-\mu f_2 - 2\alpha\mu g_2 + c\mu\delta_2 + 4a\alpha\mu\delta_2 - 2a\beta^3\mu c_0\epsilon_2 + \mu\delta_2', \beta\mu g_1 - 2a\beta\mu\delta_1 - \mu\epsilon_1', \beta\mu g_2 - 2a\beta\mu\delta_2 - \mu\epsilon_2',$
$-\mu g_1\delta_1 + a\mu\delta_1^2 - \mu g_2\delta_2 + a\mu\delta_2^2 - a\beta^2\mu c_0\epsilon_1^2 - a\beta^2\mu c_0\epsilon_2^2 + \mu\kappa_1' + \mu\kappa_2', -2d\mu - 4a\alpha\mu + \mu'\}$

```
(* Remove trivial factors. The expressions correspond to (2.41)-(2.47). *)
TableForm [
  cond = If[Head[#] === Times , Select[#, Not[FreeQ[#, Derivative]] &], #] & /@ Factor[cond]]
```

$b + 2c\alpha + 4a\alpha^2 - a\beta^4 c_0 + \alpha'$
$c\beta + 4a\alpha\beta + \beta'$
$a\beta^2 + \gamma'$
$f_1 + 2\alpha g_1 - c\delta_1 - 4a\alpha\delta_1 + 2a\beta^3 c_0\epsilon_1 - \delta_1'$
$f_2 + 2\alpha g_2 - c\delta_2 - 4a\alpha\delta_2 + 2a\beta^3 c_0\epsilon_2 - \delta_2'$
$\beta g_1 - 2a\beta\delta_1 - \epsilon_1'$
$\beta g_2 - 2a\beta\delta_2 - \epsilon_2'$
$g_1\delta_1 - a\delta_1^2 + g_2\delta_2 - a\delta_2^2 + a\beta^2 c_0\epsilon_1^2 + a\beta^2 c_0\epsilon_2^2 - \kappa_1' - \kappa_2'$
$-2d\mu - 4a\alpha\mu + \mu'$

```
(* Assuming that δ1=δ2=δ etc. we get precisely (2.46). *)
Factor[cond[[-2]] /. {δ_ → δ, κ_ → κ, ϵ_ → ϵ, g_ → g}]
```

$2(g\delta - a\delta^2 + a\beta^2\epsilon^2 c_0 - \kappa')$

3.1 Airy Beams

In[20]:=
```
(* This is Equation (3.1). *)
eqn31 = I * D[ψ[x, t], t] + D[ψ[x, t], x, x]
```

Out[20]= $i\,\psi^{(0,1)}[x, t] + \psi^{(2,0)}[x, t]$

```
(* Plug in the ansatz (3.2). *)
Together[eqn31 /. ψ → Function[{x, t}, Exp[I * (x - 2 t^2 / 3) * t] * F[x - t^2]]]
```

$e^{i t\left(-\frac{2t^2}{3}+x\right)}(t^2 F[-t^2 + x] - x F[-t^2 + x] + F''[-t^2 + x])$

```
(* This reduces to the Airy equation (3.3). *)
Expand[(% / Exp[I * (x - 2 t^2 / 3) * t]) /. x → z + t^2]
```

$-z F[z] + F''[z]$

■ Explicit action of the Schrödinger group (3.5)

In[21]:=
```
(* This is the Schrödinger group (3.5). *)
args = {
      ξ → ε[0] + ((β[0] * x - 2 * β[0] * δ[0] * t) / (1 + 4 * α[0] * t)),
      τ → (β[0] ^ 2 * t) / (1 + 4 * α[0] * t) - γ[0]};
SG = Function@@ {{x, t}, Sqrt[β[0] / (1 + 4 α[0] * t)] *
            Exp[I * ((α[0] * x ^ 2 + δ[0] * x - δ[0] ^ 2 * t) / (1 + 4 * α[0] * t) + κ[0])] * χ[ξ, τ] /. args};
TraditionalForm[SG[x, t]]
```

Out[23]//TraditionalForm=

$$\sqrt{\frac{\beta(0)}{4\,\alpha(0)\,t+1}}\,\exp\left(i\left(\kappa(0)+\frac{-\delta(0)^2\,t+\alpha(0)\,x^2+\delta(0)\,x}{4\,\alpha(0)\,t+1}\right)\right)\chi\left(\varepsilon(0)+\frac{\beta(0)\,x-2\,\beta(0)\,\delta(0)\,t}{4\,\alpha(0)\,t+1},\ \frac{\beta(0)^2\,t}{4\,\alpha(0)\,t+1}-\gamma(0)\right)$$

```
(* Plug (3.5) into (3.1) and divide out some content. *)
Together[(eqn31/.ψ→SG)/(SG[x, t]/.χ[__]→1)]
```

$$\left(\beta[0]^2\left(\mathrm{i}\,\chi^{(0,1)}\left[\frac{x\beta[0]-2t\beta[0]\delta[0]}{1+4t\alpha[0]}+\varepsilon[0],\frac{t\beta[0]^2}{1+4t\alpha[0]}-\gamma[0]\right]+\chi^{(2,0)}\left[\frac{x\beta[0]-2t\beta[0]\delta[0]}{1+4t\alpha[0]}+\varepsilon[0],\frac{t\beta[0]^2}{1+4t\alpha[0]}-\gamma[0]\right]\right)\right)\Big/(1+4t\alpha[0])^2$$

```
(* Write this more nicely. We get the second equation of (3.4). *)
%*((1+4t*α[0])/β[0])^2/.(Reverse/@args)
```

$$\mathrm{i}\,\chi^{(0,1)}[\xi,\tau]+\chi^{(2,0)}[\xi,\tau]$$

■ Multi-parameter Airy beams (3.6)

In[24]:=
```
MPAB[x_, t_] := (1+4*α*t)^(-1/2)*Exp[I*((α*x^2+δ*x-δ^2*t)/(1+4*α*t))]*
    Exp[I*(ε+((β*x-2*β*δ*t)/(1+4*α*t))-(2*(β^4*t^2)/(3*(1+4*α*t)^2)))*
      (((β^2*t)/(1+4*α*t)))]*
    AiryAi[ε+((β*x-2*β*δ*t)/(1+4*α*t))-((β^4*t^2)/((1+4*α*t)^2))];
TraditionalForm[MPAB[x, t]]
```

Out[25]//TraditionalForm=

$$\left(\exp\left(\left(i\,\beta^2\,t\left(\varepsilon-\frac{2\,\beta^4\,t^2}{3\,(4\,\alpha\,t+1)^2}+\frac{\beta\,x-2\,\beta\,\delta\,t}{4\,\alpha\,t+1}\right)\right)\Big/(4\,\alpha\,t+1)+\frac{i\left(-\delta^2\,t+\alpha\,x^2+\delta\,x\right)}{4\,\alpha\,t+1}\right)\mathrm{Ai}\left(-\frac{t^2\,\beta^4}{(4\,t\,\alpha+1)^2}+\frac{x\,\beta-2\,t\,\beta\,\delta}{4\,t\,\alpha+1}+\varepsilon\right)\right)\Big/\left(\sqrt{4\,\alpha\,t+1}\right)$$

```
Timing[FullSimplify[eqn31/.ψ→MPAB]]
```

{0.172, 0}

Now we derive the Schrödinger equation (3.1) from the definition (3.6), using the HolonomicFunctions package.
At the same time, this proves that (3.6) solves (3.1).

```
(* This computes a system of PDEs w.r.t. t and x. *)
Timing[ann=Annihilator[MPAB[x, t], {Der[t], Der[x]}]]
```

$\{0.172, \{(1+8t\alpha+16t^2\alpha^2)D_t+(4x\alpha+16tx\alpha^2+2t\beta^3+2\delta+8t\alpha\delta)D_x+(2\alpha+8t\alpha^2-4\mathrm{i}x^2\alpha^2-\mathrm{i}x\beta^3-4\mathrm{i}x\alpha\delta-\mathrm{i}\delta^2-\mathrm{i}\beta^2\varepsilon),\ (1+8t\alpha+16t^2\alpha^2)D_x^2+(-4\mathrm{i}x\alpha-16\mathrm{i}tx\alpha^2-2\mathrm{i}t\beta^3-2\mathrm{i}\delta-8\mathrm{i}t\alpha\delta)D_x+(-2\mathrm{i}\alpha-8\mathrm{i}t\alpha^2-4x^2\alpha^2-x\beta^3-4x\alpha\delta-\delta^2-\beta^2\varepsilon)\}\}$

```
(* This finds a linear combination of the
  above equations such that only B_t and B_xx occur. *)
FindRelation[ann, Support→{Der[t], Der[x]^2}]
```

$\{D_x^2+\mathrm{i}\,D_t\}$

■ Graphical example of “self-accelerating” Airy mode in Equation (3.6)

```
(* Special case of the Airy mode in Equation (3.6) when α=-1/4, β=1, δ=ε=0. *)
PowerExpand[ComplexExpand [Abs[MPAB[x, t]/.{t→s, α→-1/4, β→1, δ→0, ε→0}]]]
```

$$\frac{\mathrm{AiryAi}\left[-\frac{s^2}{(1-s)^2}+\frac{x}{1-s}\right]}{\sqrt{1-s}}$$

```
(* Figure 1. Levels of intensity of the Airy mode under consideration *)
ContourPlot[%^2, {x, -10, 5}, {s, -3, .5}, PlotPoints→50, Axes→True, AxesLabel→{x, s}]
```



■ Quasi-diffraction-free finite energy Airy beams

In[82]:=
```
(* Let us consider another special case of the Airy beam in Equation (3.6). *)
FEAB0 = AiryAi[z] / Sqrt[γ - t] * Exp[w] /.
   {z→g^(1/3) * (-g * (γ * t)^2 / (γ - t)^2 + (γ * (x - v * t) / (γ - t))),
    w→I * ((x - v * t / 2) * (v * γ)) / (2 * (γ - t)) + I * g * (x - v * t) * (γ^2 * t) / (γ - t)^2}
```

Out[82]= $\frac{1}{\sqrt{-t+\gamma}} e^{\frac{i g t (-t v+x)\gamma^2}{(-t+\gamma)^2}+\frac{i v \left(-\frac{t v}{2}+x\right)\gamma}{2(-t+\gamma)}} \text{AiryAi}\left[g^{1/3}\left(-\frac{g t^2 \gamma^2}{(-t+\gamma)^2}+\frac{(-t v+x)\gamma}{-t+\gamma}\right)\right]$

In[101]:= $\text{FEAB} = \text{FEAB0} * \text{Exp}\left[\left(i\left(3 t^2 x^2 - 6 t x^2 \gamma + 3 x^2 \gamma^2 + 8 g^2 t^3 \gamma^3\right)\right) / \left(12 (t-\gamma)^3\right)\right]$

Out[101]= $\frac{1}{\sqrt{-t+\gamma}} e^{\frac{i g t (-t v+x)\gamma^2}{(-t+\gamma)^2}+\frac{i v \left(-\frac{t v}{2}+x\right)\gamma}{2(-t+\gamma)}+\frac{i\left(3 t^2 x^2-6 t x^2\gamma+3 x^2\gamma^2+8 g^2 t^3\gamma^3\right)}{12 (t-\gamma)^3}} \text{AiryAi}\left[g^{1/3}\left(-\frac{g t^2 \gamma^2}{(-t+\gamma)^2}+\frac{(-t v+x)\gamma}{-t+\gamma}\right)\right]$

In[84]:=
```
(* Let us check that it satisfies the equation. *)
FullSimplify[I * D[FEAB, t] + D[FEAB, x, x]]
```

Out[84]= 0

In[102]:=
```
(* Compute the equation, from the expression, using HolonomicFunctions. *)
FindRelation[Annihilator[FEAB, {Der[t], Der[x]}], Support→{Der[t], Der[x]^2}]
```

Out[102]= $\{D_x^2 + i D_t\}$

In[86]:=
```
(* Let us choose the following values of the parameters . *)
FEAB = FEAB /. {v→-I / 10, g→1, γ→1};
```

In[94]:=
```
(* Plotting at t=0. *)
Plot[Abs[FEAB/.t→0]^2, {x, -15, 3}, AxesLabel→{x, Abs[ψ]^2}]
```

Out[94]=



In[95]:=
```
(* Plotting at t=0.5. *)
Plot[Abs[FEAB/.t→1/2]^2, {x, -5, 2}, PlotStyle→Blue, AxesLabel→{x, Abs[ψ]^2}]
```

Out[95]=



In[96]:=
```
(* Showing plots at t=0 and t=0.5 together. *)
Show[%, %%, PlotRange→{{-5, 2}, {0, 1/2}}]
```

Out[96]=



In[100]:=
```
(* "Rogue wave" animation *)
Animate[
  Plot[Abs[FEAB/.t→tt]^2, {x, -10, 6},
   PlotRange→{0, 0.57}, AxesLabel→{x, Abs[ψ]^2}], {tt, 0, .95}]
```

3.2 Oscillating and Breathing Hermite-Gaussian Beams

In[26]:=
```
(* This is Equation (3.7). *)
eqn37=2*I*D[A[x, t], t]+D[A[x, t], x, x]-(x^2)*A[x, t]
```

Out[26]= $-x^2 A[x, t] + 2 i A^{(0,1)}[x, t] + A^{(2,0)}[x, t]$

■ Standard harmonic oscillator solutions

In[27]:=
```
SHO[x_, t_] := (e^(-x^2/2) HermiteH[n, x])/((2^n n! Sqrt[π])^(1/2)) * Exp[-I*(n+1/2)*t];
TraditionalForm[SHO[x, t]]
```

Out[28]//TraditionalForm=

$$\frac{H_n(x)\, e^{-\frac{x^2}{2}-i\left(n+\frac{1}{2}\right)t}}{\sqrt[4]{\pi}\,\sqrt{2^n\, n!}}$$

```
TraditionalForm[Table[SHO[x, t], {n, 0, 2}]]
```

$$\left\{\frac{e^{-\frac{x^2}{2}-\frac{i t}{2}}}{\sqrt[4]{\pi}}, \frac{\sqrt{2}\, x\, e^{-\frac{x^2}{2}-\frac{3 i t}{2}}}{\sqrt[4]{\pi}}, \frac{(4x^2-2)\, e^{-\frac{x^2}{2}-\frac{5 i t}{2}}}{2\sqrt{2}\,\sqrt[4]{\pi}}\right\}$$

```
Timing[FullSimplify[eqn37/.A→SHO]]
```

{0.172, 0}

■ “Missing” solutions (3.8) - (3.14)

In[29]:=
```
MS[x_, t_] := (HermiteH[n, β[t]*x+ε[t]])/((μ[t]*2^n n! Sqrt[π])^(1/2)) * Exp[-(β[t]*x+ε[t])^2/2]*
    Exp[I*(α[t]*x^2+δ[t]*x+κ[t])]*Exp[I*(2*n+1)*γ[t]];
TraditionalForm[
  MS[
   x,
   t]]
```

Out[30]//TraditionalForm=

$$\left(H_n(\varepsilon(t)+x\,\beta(t))\,\exp\!\left(i\,(2n+1)\,\gamma(t)+i\left(\kappa(t)+x^2\,\alpha(t)+x\,\delta(t)\right)-\frac{1}{2}\,(\varepsilon(t)+x\,\beta(t))^2\right)\right)\Big/\left(\sqrt[4]{\pi}\,\sqrt{2^n\, n!\,\mu(t)}\right)$$

In[31]:=
```
(* These are Equations (3.9) - (3.14). *)
eqns39to314 = {μ[t] → μ * Sqrt[β^4 * Sin[t]^2 + (2 * α * Sin[t] + Cos[t])^2],
    α[t] → (α * Cos[2 * t] + 1 / 4 * Sin[2 * t] * (β^4 + 4 * α^2 - 1)) /
        (β^4 * Sin[t]^2 + (2 * α * Sin[t] + Cos[t])^2),
    β[t] → β / Sqrt[(β^4 * Sin[t]^2 + (2 * α * Sin[t] + Cos[t])^2)],
    γ[t] → γ - (1 / 2) * ArcTan[β^2 * Sin[t] / (2 * α * Sin[t] + Cos[t])],
    δ[t] → (δ * (2 * α * Sin[t] + Cos[t]) + ε * β^3 * Sin[t]) /
        (β^4 * (Sin[t]^2) + (2 * α * Sin[t] + Cos[t])^2),
    ε[t] → (ε * (2 * α * Sin[t] + Cos[t]) - β * δ * Sin[t]) /
        Sqrt[β^4 * Sin[t]^2 + (2 * α * Sin[t] + Cos[t])^2],
    κ[t] → κ + Sin[t]^2 * (ε * β^2 * (α * ε - β * δ) - α * δ^2) /
            (β^4 * (Sin[t])^2 + (2 * α * Sin[t] + Cos[t])^2) +
        (1 / 4) * Sin[2 * t] * (ε^2 * β^2 - δ^2) / (β^4 * Sin[t]^2 + (2 * α * Sin[t] + Cos[t])^2)};
TraditionalForm[TableForm [eqns39to314]]
```

Out[32]//TraditionalForm=

$$\mu(t) \to \mu \sqrt{(2\,\alpha \sin(t) + \cos(t))^2 + \beta^4 \sin^2(t)}$$
$$\alpha(t) \to \frac{\frac{1}{4}\left(4\,\alpha^2+\beta^4-1\right)\sin(2\,t)+\alpha\cos(2\,t)}{(2\,\alpha\sin(t)+\cos(t))^2+\beta^4\sin^2(t)}$$
$$\beta(t) \to \frac{\beta}{\sqrt{(2\,\alpha\sin(t)+\cos(t))^2+\beta^4\sin^2(t)}}$$
$$\gamma(t) \to \gamma - \frac{1}{2}\tan^{-1}\left(\frac{\beta^2\sin(t)}{2\,\alpha\sin(t)+\cos(t)}\right)$$
$$\delta(t) \to \frac{\delta(2\,\alpha\sin(t)+\cos(t))+\beta^3\,\varepsilon\sin(t)}{(2\,\alpha\sin(t)+\cos(t))^2+\beta^4\sin^2(t)}$$
$$\varepsilon(t) \to \frac{\varepsilon(2\,\alpha\sin(t)+\cos(t))-\beta\,\delta\sin(t)}{\sqrt{(2\,\alpha\sin(t)+\cos(t))^2+\beta^4\sin^2(t)}}$$
$$\kappa(t) \to \kappa + \frac{\sin(2\,t)\left(\beta^2\,\varepsilon^2-\delta^2\right)}{4\left((2\,\alpha\sin(t)+\cos(t))^2+\beta^4\sin^2(t)\right)} + \frac{\sin^2(t)\left(\beta^2\,\varepsilon\,(\alpha\,\varepsilon-\beta\,\delta)-\alpha\,\delta^2\right)}{(2\,\alpha\sin(t)+\cos(t))^2+\beta^4\sin^2(t)}$$

```
Timing[FullSimplify[eqn37 /. A → Function @@ {{x, t}, MS[x, t] /. eqns39to314}]]
```

```
{229.914, 0}
```

```
(* With a little effort one can reduce the simplification time drastically. *)
Timing[
  (* The exponential part. *)
  partExp =
   Exp[-(β[t] * x + ε[t])^2 / 2] * Exp[I * (α[t] * x^2 + δ[t] * x + κ[t])] * Exp[I * ((2 * n + 1) * γ[t])];
  (* The HermiteH part
    (we replace HermiteH by H in order to avoid automatic evaluation). *)
  partH = H[n, β[t] * x + ε[t]];
  (* Rewrite Sin[2t] and Cos[2t]. *)
  {partExp, partH} = {partExp, partH} /. eqns39to314 /.
     {Sin[2 t] → 2 Sin[t] Cos[t], Cos[2 t] → Cos[t]^2 - Sin[t]^2};
  (* Plug the expression into (3.1). *)
  test = eqn37 /.
     A → Function @@ {{x, t}, partExp * partH / Sqrt[μ[t] * 2^n * n! * Sqrt[Pi]] /. eqns39to314};
  (* Now we have to simplify the 2nd derivative of HermiteH by hand. *)
  test = test /. Derivative[0, 2][H][n_, x_] :→ 2 * x * Derivative[0, 1][H][n, x] - 2 * n * H[n, x];
  (* We divide by the exponential part and clear denominators . *)
  test = Numerator [Together[test / partExp]];
  (* Consider the coefficients of H and H' separately and simplify. *)
  Simplify[Coefficient[test, {partH, Derivative[0, 1][H] @@ partH}]]
]
```

```
{1.887, {0, 0}}
```

▪ Deriving equation (3.7) using the substitution t → ⅈ Log[z]

```
(* We split (3.8) into two factors. *)
{expr1, expr2} = {(μ[t] * 2^n * n! * Sqrt[Pi])^(-1/2) *
        HermiteH[n, β[t] * x + ε[t]] * Exp[-(β[t] * x + ε[t])^2/2] *
        Exp[I * (α[t] * x^2 + δ[t] * x + κ[t])], Exp[I * (2 * n + 1) * γ[t]]};
MS[x, t] === expr1 * expr2
```

```
True
```

```
(* Now the substitutions; the last one (t → I*Log[z]) has the
  advantage that Sin[t] and Cos[t] turn into rational functions in z. *)
{expr1, expr2} = {expr1, expr2} /. eqns39to314 /.
          {Sin[2 t] → 2 Sin[t] Cos[t], Cos[2 t] → Cos[t]^2 - Sin[t]^2} /.
        t → I * Log[z] /. (a : (Sin|Cos)[_] :→ FunctionExpand[a]);
expr2
```

$$e^{i(1+2n)\left(\gamma - \frac{1}{2} i \,\mathrm{ArcTanh}\left[\frac{(-1+z^2)\beta^2}{2z\left(\frac{1+z^2}{2z} + \frac{i(-1+z^2)\alpha}{z}\right)}\right]\right)}$$

```
(* Remove  constant factor that confuses Annihilator here;
it doesn't change the result. *)
expr2 = expr2 /. γ → 0
```

$$e^{\frac{1}{2}(1+2n)\,\mathrm{ArcTanh}\left[\frac{(-1+z^2)\beta^2}{2z\left(\frac{1+z^2}{2z} + \frac{i(-1+z^2)\alpha}{z}\right)}\right]}$$

```
Timing[ann = Annihilator[expr1 * expr2, {Der[x], Der[z]}]]
```

A very large output was generated. Showing a sample of it.

$\{9.048, \{(2x - 2xz^4 - 8ix\alpha - 8ixz^4\alpha - 8x\alpha^2 + 8xz^4\alpha^2 - 2x\beta^4 + 2xz^4\beta^4 - 4iz\delta - 4iz^3\delta - 8z\alpha\delta + 8z^3\alpha\delta - 4z\beta^3\varepsilon + 4z^3\beta^3\varepsilon) D_x + (-2z - 4z^3 - 2z^5 + 8iz\alpha - 8iz^5\alpha + \ll 1 \gg - 16z^3\alpha^2 + 8z^5\alpha^2 + 2z\beta^4 - 4z^3\beta^4 + 2z^5\beta^4) D_z + (1 + 2x^2 - z^4 + \ll 39 \gg), \ll 1 \gg\}\}$

show less | show more | show all | set size limit ...

```
(* This is indeed very large. *)
ByteCount /@ ann
```

```
{12704, 737064}
```

```
(* Find an operator with the desired support in the computed  left ideal. *)
Timing[rel = FindRelation[ann, Support → {Der[x]^2, Der[z], 1}]]
```

$\{1.155, \{D_x^2 + 2zD_z - x^2\}\}$

```
(* This is equation (3.7). *)
ApplyOreOperator[First[rel], A[x, I * Log[z]]] /. I * Log[z] → s
```

$-x^2 A[x, s] + 2 i A^{(0,1)}[x, s] + A^{(2,0)}[x, s]$

3.3 Hermite-Gaussian Beams

In[33]:= `eqn315 = 2 * I * D[B[x, s], s] + D[B[x, s], x, x]`

Out[33]= $2 i B^{(0,1)}[x, s] + B^{(2,0)}[x, s]$

■ Deriving equation (3.7) from (3.15) and (3.16)

```
(* Plug the substitution (3.16) into equation (3.15): *)
transEqn = eqn315 /. B → Function[{x, s},
         Exp[I*s*x^2/2/(1+s^2)]/(1+s^2)^(1/4)*A[x/Sqrt[1+s^2], ArcTan[s]]];
(* Divideby the exponential factor and clear denominators . *)
transEqn = Numerator [Together[transEqn / Exp[I*s*x^2/2/(1+s^2)]]]
```

$$-x^2 A\left[\frac{x}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] + 2\,i\,A^{(0,1)}\left[\frac{x}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] + 2\,i\,s^2 A^{(0,1)}\left[\frac{x}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] + A^{(2,0)}\left[\frac{x}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] + s^2 A^{(2,0)}\left[\frac{x}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right]$$

```
(* Now substitute the variables x and s in order to get (3.7). *)
Together[(transEqn /. {x → x*Sqrt[1+s^2], ArcTan[s] → s}) / (1+s^2)]
```

$$-x^2 A[x, s] + 2\,i\,A^{(0,1)}[x, s] + A^{(2,0)}[x, s]$$

```
Simplify[% == eqn37 /. t → s]
```

True

■ Hermite-Gaussian modes (3.17) for the homogeneous parabolic equation

In[34]:=
```
(* If gamma  is set to  0, then this is Equation (3.17). *)
```

$$\mathrm{HGP}[x\_, s\_] := \frac{1}{\pi^{1/4}\sqrt{2^n\, n!}}\, e^{\frac{i\,(-s\,\delta^2+2\,x\,(x\,\alpha+\delta))-(x\,\beta+\epsilon)^2-2\,s\,\epsilon\,(-\beta\,\delta+\alpha\,\epsilon)}{2\,(1+2\,s\,\alpha+i\,s\,\beta^2)}+i\,(1+2\,n)\left(\gamma-\frac{1}{2}\mathrm{ArcTan}\left[\frac{s\,\beta^2}{1+2\,s\,\alpha}\right]\right)} \sqrt{\frac{\beta}{\sqrt{(1+2\,s\,\alpha)^2+s^2\,\beta^4}}}\ \mathrm{HermiteH}\left[n, \frac{x\,\beta-s\,\beta\,\delta+\epsilon+2\,s\,\alpha\,\epsilon}{\sqrt{(1+2\,s\,\alpha)^2+s^2\,\beta^4}}\right];$$

```
TraditionalForm[HGP[x, s]]
```

Out[35]//TraditionalForm=

$$\left(\sqrt{\frac{\beta}{\sqrt{\beta^4\,s^2+(2\,\alpha\,s+1)^2}}}\ H_n\!\left(\frac{\epsilon+2\,\alpha\,\epsilon\,s-\beta\,\delta\,s+\beta\,x}{\sqrt{\beta^4\,s^2+(2\,\alpha\,s+1)^2}}\right)\exp\!\left(i\,(2\,n+1)\left(\gamma-\frac{1}{2}\tan^{-1}\!\left(\frac{\beta^2\,s}{2\,\alpha\,s+1}\right)\right)+\left(-2\,\epsilon\,s\,(\alpha\,\epsilon-\beta\,\delta)+i\left(2\,x\,(\delta+\alpha\,x)-\delta^2\,s\right)-(\epsilon+\beta\,x)^2\right)\Big/\left(2\left(2\,\alpha\,s+i\,\beta^2\,s+1\right)\right)\right)\right)\Big/\left(\sqrt[4]{\pi}\ \sqrt{2^n\,n!}\right)$$

```
Timing[FullSimplify[eqn315 /. B → HGP]]
```

{5.428, 0}

```
(* This computes  a system  of PDEs for chi w.r.t. s and x. *)
Timing[ann = Annihilator[HGP[x, s], {Der[s], Der[x]}]]
```

$$\{0.281, \{(2+8\,s\,\alpha+8\,s^2\,\alpha^2+2\,s^2\,\beta^4)\,D_s+(4\,x\,\alpha+8\,s\,x\,\alpha^2+2\,s\,x\,\beta^4+2\,\delta+4\,s\,\alpha\,\delta+2\,s\,\beta^3\,\epsilon)\,D_x+(2\,\alpha+4\,s\,\alpha^2-4\,i\,x^2\,\alpha^2+i\,\beta^2+2\,i\,n\,\beta^2+s\,\beta^4-i\,x^2\,\beta^4-4\,i\,x\,\alpha\,\delta-i\,\delta^2-2\,i\,x\,\beta^3\,\epsilon-i\,\beta^2\,\epsilon^2),$$
$$(1+4\,s\,\alpha+4\,s^2\,\alpha^2+s^2\,\beta^4)\,D_x^2+(-4\,i\,x\,\alpha-8\,i\,s\,x\,\alpha^2-2\,i\,s\,x\,\beta^4-2\,i\,\delta-4\,i\,s\,\alpha\,\delta-2\,i\,s\,\beta^3\,\epsilon)\,D_x+(-2\,i\,\alpha-4\,i\,s\,\alpha^2-4\,x^2\,\alpha^2+\beta^2+2\,n\,\beta^2-i\,s\,\beta^4-x^2\,\beta^4-4\,x\,\alpha\,\delta-\delta^2-2\,x\,\beta^3\,\epsilon-\beta^2\,\epsilon^2)\}\}$$

```
(* This finds a linear combination of the above
  equations such that only chi_s and chi_xx occur. *)
FindRelation[ann, Support → {Der[s], Der[x]^2}]
```

$$\{D_x^2+2\,i\,D_s\}$$

Graphical example of 1D self-focusing Gaussian mode in Equation (3.17)

```
(* This is intensity,
or the modulus squared of the Gaussian mode for n=0 in Equation (3.17). *)
Clear[ABS2HGP0];
SetDelayed@@{ABS2HGP0[x_, s_, α_, β_, δ_, ε_],
      Simplify[ComplexExpand [Abs[HGP[x, s]]^2 /. {n→0, γ→0}]]};
TraditionalForm[ABS2HGP0[x, s, α, β, δ, ε]]
```

$$\frac{\sqrt{\beta^2}\ \exp\left(-\frac{(\varepsilon+2\,\alpha\,\varepsilon\,s-\beta\,\delta\,s+\beta\,x)^2}{s^2\,(4\,\alpha^2+\beta^4)+4\,\alpha\,s+1}\right)}{\sqrt{\pi}\ \sqrt{s^2\,(4\,\alpha^2+\beta^4)+4\,\alpha\,s+1}}$$

```
(* Compute the location of the maximum . *)
{max } = Solve[Thread[ (D[ABS2HGP0[x, s, α, β, δ, ε], #] & /@ {x, s}) == 0] /. Exp[a_] → 1, {x, s}]
```

$$\left\{\left\{x\to\frac{-2\,\alpha\,\delta-\beta^3\,\varepsilon}{4\,\alpha^2+\beta^4},\ s\to-\frac{2\,\alpha}{4\,\alpha^2+\beta^4}\right\}\right\}$$

```
(* The maximum is given by *)
Simplify[ABS2HGP0[x, s, α, β, δ, ε] /.max , Element [β, Reals]]
```

$$\frac{\mathrm{Abs}[\beta]}{\sqrt{\pi}\ \beta^2\ \sqrt{\frac{1}{4\,\alpha^2+\beta^4}}}$$

```
(* Choosing special values of the parameters as follows: α=-2, β=1, δ=ε=0. *)
ABS2HGP0[x, s, -2, 1, 0, 0]
```

$$\frac{e^{-\frac{x^2}{1-8\,s+17\,s^2}}}{\sqrt{\pi}\ \sqrt{1-8\,s+17\,s^2}}$$

```
max /. {α→-2, β→1, δ→0, ε→0}
```

$$\left\{x\to 0,\ s\to\frac{4}{17}\right\}$$

```
(* The value of the maximum for the given parameters . *)
%% /. %
```

$$\sqrt{\frac{17}{\pi}}$$

```
N[%]
```

2.32621

```
(* Graph of intensity for the Gaussian mode  under consideration when α=-2,
β=1, δ=ε=0. *) Plot3D[ABS2HGP0[x, s, -2, 1, 0, 0], {x, -1, 1}, {s, -.105, .575},
  PlotPoints→20, PlotRange→{0, 2.33}, AxesLabel→{x, s, "Intensity"}]
```




```
(* The corresponding intensity levels. *) ContourPlot[
  ABS2HGP0[x, s, -2, 1, 0, 0], {x, -1, 1}, {s, -.105, .575}, Axes→True, AxesLabel→{x, s}]
```



■ Elegant Hermite-Gaussian modes (3.19)

We derive Equation (3.19) from (3.17) by means of the substitution given in (3.18).

```
(* Perform  the substitution of (3.18) into (3.17). *)
expr = HGP[x, s] /. γ → 0 /.
    ((1 + 2 α s) ^ 2 + β ^ 4 s ^ 2) ^ e_ → ((1 + 2 α s + I * β ^ 2 s) / Exp[I * ArcTan[β ^ 2 s / (1 + 2 α s)]]) ^ (2 e) /.
  α → I * β ^ 2 / 2
```

$$\frac{1}{\pi^{1/4}\sqrt{2^n n!}} e^{\frac{i\left(-s\delta^2+2x\left(\frac{1}{2}i x\beta^2+\delta\right)\right)-(x\beta+\varepsilon)^2-2s\varepsilon\left(-\beta\delta+\frac{1}{2}i\beta^2\varepsilon\right)}{2\left(1+2is\beta^2\right)}-\frac{1}{2}i(1+2n)\,\mathrm{ArcTan}\left[\frac{s\beta^2}{1+is\beta^2}\right]}$$

$$\sqrt{\frac{e^{i\,\mathrm{ArcTan}\left[\frac{s\beta^2}{1+is\beta^2}\right]}\beta}{1+2is\beta^2}}\ \mathrm{HermiteH}\left[n, \left(e^{i\,\mathrm{ArcTan}\left[\frac{s\beta^2}{1+is\beta^2}\right]}(x\beta-s\beta\delta+\varepsilon+is\beta^2\varepsilon)\right)\Big/(1+2is\beta^2)\right]$$

```
(* Express the arctan by logarithms  for further simplifications *)
expr = Simplify[expr /. ArcTan[x_] → I / 2 * (Log[1 - I * x] - Log[1 + I * x])]
```

$$\frac{1}{\pi^{1/4}\sqrt{2^n n!}} e^{\frac{i\left(2x^2\beta^2+\varepsilon^2+2x(-i\delta+\beta\varepsilon)+is\left(\delta^2+2i\beta\delta\varepsilon+\beta^2\varepsilon^2\right)\right)}{-2i+4s\beta^2}}\left(-\frac{i}{-i+s\beta^2}\right)^{\frac{1}{4}+\frac{n}{2}}\sqrt{\frac{\beta\sqrt{\frac{-1-2is\beta^2}{(-i+s\beta^2)^2}}\,(-i+s\beta^2)}{-i+2s\beta^2}}\left(\frac{-i+2s\beta^2}{-i+s\beta^2}\right)^{\frac{1}{4}(-1-2n)}$$

$$\mathrm{HermiteH}\left[n, -\left(\left(i\sqrt{1+\frac{s\beta^2}{-i+s\beta^2}}\,(x\beta+\varepsilon+s\beta(-\delta+i\beta\varepsilon))\right)\Big/\left(\sqrt{-\frac{i}{-i+s\beta^2}}\,(-i+2s\beta^2)\right)\right)\right]$$

```
(* Now simplify the square-roots. *)
(* CAVEAT: PowerExpand simplifies Sqrt[x^2] to x. *)
expr = PowerExpand[expr /. Power[x_, a_] :→ Power[Factor[x], a]]
```

$$\frac{1}{\pi^{1/4}\sqrt{n!}}(-i)^{\frac{1}{4}+\frac{n}{2}}2^{-n/2}e^{\frac{i\left(2x^2\beta^2+\varepsilon^2+2x(-i\delta+\beta\varepsilon)+is\left(\delta^2+2i\beta\delta\varepsilon+\beta^2\varepsilon^2\right)\right)}{-2i+4s\beta^2}}\sqrt{\beta}\,(-1-2is\beta^2)^{1/4}$$

$$(-i+s\beta^2)^{-\frac{1}{4}-\frac{n}{2}+\frac{1}{4}(1+2n)}(-i+2s\beta^2)^{-\frac{1}{2}+\frac{1}{4}(-1-2n)}\,\mathrm{HermiteH}\left[n, -\frac{(-1)^{3/4}(x\beta+\varepsilon+s\beta(-\delta+i\beta\varepsilon))}{\sqrt{-i+2s\beta^2}}\right]$$

```
(* Enforce that all Sqrt's have argument  1+2isβ^2. *)
expr = Simplify[expr /. Power[x_ /; FreeQ[Together[x / (1 + 2 * I * β ^ 2 * s)], s], a_] :→
      Power[Together[x / (1 + 2 * I * β ^ 2 * s)], a] * Power[1 + 2 * I * β ^ 2 * s, a]]
```

$$\frac{1}{\pi^{1/4}\sqrt{n!}}i\,2^{-n/2}e^{\frac{i\left(2x^2\beta^2+\varepsilon^2+2x(-i\delta+\beta\varepsilon)+is\left(\delta^2+2i\beta\delta\varepsilon+\beta^2\varepsilon^2\right)\right)}{-2i+4s\beta^2}}\sqrt{\beta}\,(1+2is\beta^2)^{\frac{1}{2}(-1-n)}\,\mathrm{HermiteH}\left[n, \frac{x\beta+\varepsilon+s\beta(-\delta+i\beta\varepsilon)}{\sqrt{1+2is\beta^2}}\right]$$

In[36]:=
```
(* This is Equation (3.19) as given in the paper. *)
EH[x_, s_] :=
  Sqrt[β / (2^n * n! (1 + 2 i s β^2) ^ (n + 1) Sqrt[Pi])] * HermiteH[n, (β (x - s δ + i s β ε) + ε) / Sqrt[1 + 2 i s β^2]] *
    Exp[-((2 β^2 x^2 - i δ (2 x - s δ) + 2 β ε (x - s δ) + (1 + i s β^2) ε^2) / (2 (1 + 2 i s β^2)))];
TraditionalForm[
 EH[
  x,
  s]]
```

Out[37]//TraditionalForm=

$$\frac{1}{\sqrt[4]{\pi}}\sqrt{\frac{\beta\,2^{-n}\left(1+2i\beta^2 s\right)^{-n-1}}{n!}}$$

$$\exp\left(-\left(\left(\varepsilon^2\left(1+i\beta^2 s\right)+2\beta\varepsilon(x-\delta s)-i\delta(2x-\delta s)+2\beta^2x^2\right)\Big/\left(2\left(1+2i\beta^2 s\right)\right)\right)\right)H_n\left(\frac{\varepsilon+\beta(i\beta\varepsilon s-\delta s+x)}{\sqrt{1+2i\beta^2 s}}\right)$$

```
(* Verify that the above derivation agrees with (3.19). *)
Simplify[PowerExpand[expr / EH[x, s]]]
```

i

```
Timing[FullSimplify[eqn315 /. B→EH]]
```

{2.075, 0}

```
(* This computes a system of PDEs for EH w.r.t. s and x. *)
Timing[ann = Annihilator[EH[x, s], {Der[s], Der[x]}]]
```

$\{0.234, \{(-2 i + 4 s \beta^2) D_s + (2 x \beta^2 - 2 i \delta + 2 s \beta^2 \delta - 2 i s \beta^3 \varepsilon) D_x + (2 \beta^2 + 2 n \beta^2 - 2 i x \beta^2 \delta - \delta^2 - 2 x \beta^3 \varepsilon - \beta^2 \varepsilon^2),$
$(-i + 2 s \beta^2) D_x^2 + (-2 i x \beta^2 - 2 \delta - 2 i s \beta^2 \delta - 2 s \beta^3 \varepsilon) D_x +$
$(-2 i \beta^2 - 2 i n \beta^2 - 2 x \beta^2 \delta + i \delta^2 + 2 i x \beta^3 \varepsilon + i \beta^2 \varepsilon^2)\}\}$

```
(* This finds a linear combination of the
  above equations such that only EH_s and EH_xx occur. *)
FindRelation[ann, Support→{Der[s], Der[x]^2}]
```

$\{-i D_x^2 + 2 D_s\}$

■ Graphical example of change in elegant Gaussian mode direction of propagation

In[38]:=
```
(* This is a renormalized version of Equation (3.20). *)
```

$$\text{ABS2EG}[x\_, s\_, \delta\_, \varepsilon\_] := \sqrt{\frac{1}{1 + 4 s^2 \beta^4}}\, e^{-\frac{2 (\beta^2 (x - s\delta)^2 + \beta (x - s\delta) \varepsilon + (1 + 3 s^2 \beta^4) \varepsilon^2)}{1 + 4 s^2 \beta^4}} \;/.\; \beta \to 1 / \text{Sqrt}[2];$$

```
TraditionalForm[ABS2EG[x, s, δ, ε]]
```

Out[39]//TraditionalForm=

$$\sqrt{\frac{1}{s^2 + 1}}\ \exp\left(-\frac{1}{s^2 + 1} 2 \left(\varepsilon^2 \left(\frac{3 s^2}{4} + 1\right) + \frac{\varepsilon (x - \delta s)}{\sqrt{2}} + \frac{1}{2} (x - \delta s)^2\right)\right)$$

```
ABS2EG[x, s, 0, 0]
```

$$e^{-\frac{x^2}{1 + s^2}} \sqrt{\frac{1}{1 + s^2}}$$

```
(* Graph of intensity for the given Gaussian mode from Equation (3.20). *)
Plot3D[ABS2EG[x, s, 0, 0], {x, -2, 2}, {s, -2.3, 2.3},
  PlotPoints→20, PlotRange→{0, 1}, AxesLabel→{x, s, "Intensity"}]
```



```
(* Levels of intensity from the previous graph. *)
ContourPlot[ABS2EG[x, s, 0, 0], {x, -2, 2}, {s, -2.3, 2.3}, Axes→True, AxesLabel→{x, s}]
```




```
(* Another choice of parameters . *)
FullSimplify[ABS2EG[x, s, 1/2, 1/2]]
```

$$e^{-\frac{4+5s^2+4x\left(\sqrt{2}+2x\right)-2s\left(\sqrt{2}+4x\right)}{8\left(1+s^2\right)}}\sqrt{\frac{1}{1+s^2}}$$

```
(* Example of a slanted beam propagation. *)
Plot3D[ABS2EG[x, s, 1/2, 1/2], {x, -4, 3.5}, {s, -2.1, 2.1},
  PlotPoints→50, PlotRange→{0, 1}, AxesLabel→{x, s, "Intensity"}]
```



```
(* Figure 2. Levels of intensity from the previous graph. *)ContourPlot[
 ABS2EG[x, s, 1/2, 1/2], {x, -4, 3.5}, {s, -2.1, 2.1}, Axes→True, AxesLabel→{x, s}]
```



■ **Graphical example of self-focusing**

In[40]:=
```
(* This is a verificationof the expansion transformation of the Schrödinger group. *)
args = {
     X→ (x / (1 + m *s)),
     S→ s / (1 + m *s)};
ET = Function@@ {{x, s}, Sqrt[1 / (1 + m *s)] *
           Exp[I * ((m *x^2) / 2 / (1 + m *s))] * C[X, S] /. args};
TraditionalForm[ET[x, s]]
```

Out[42]//TraditionalForm=

$$\sqrt{\frac{1}{m\,s+1}}\;e^{\frac{i\,m}{2\,(m}\,\frac{x^2}{s+1)}}\;C\left[\frac{x}{m\,s+1},\,\frac{s}{m\,s+1}\right]$$

```
(* Plug the expansion into (3.15) and divideout some content. *)
Together[(eqn315 /. B→ET) / (ET[x, s] /. C[__]→1)]
```

$$\frac{1}{(1+m\ s)^2}\left(2\,i\,C^{(0,1)}\left[\frac{x}{1+m\ s},\,\frac{s}{1+m\ s}\right]+C^{(2,0)}\left[\frac{x}{1+m\ s},\,\frac{s}{1+m\ s}\right]\right)$$

```
(* Write this more nicely. We get the same equation (3.15). *)
%* (1 + m *s) ^2 /. (Reverse /@args) /. {X→x, S→s}
```

$2\,i\,C^{(0,1)}[x,\,s]+C^{(2,0)}[x,\,s]$

In[43]:=
```
(* This is the transformed  elegant Gaussian package from  the paper up to a constant. *)
EGT[x_, s_] := Exp[-((2 x^2 β^2 + 2 (1 + m s) x (-ⅈ δ + β ε) + (1 + m s) (ε^2 + s (ⅈ δ^2 - 2 β δ ε + (m + ⅈ β^2) ε^2))) /
        (2 (1 + m s) (1 + m s + 2 ⅈ s β^2)))]
    Sqrt[β / (1 + m s + 2 ⅈ s β^2)] * Exp[I * ((m * x^2) / 2 / (1 + m * s))];
TraditionalForm[EGT[x, s]]
```

Out[44]//TraditionalForm=

$$\sqrt{\frac{\beta}{m\,s+2\,i\,\beta^2\,s+1}}\;\exp\left(\frac{i\,m\,x^2}{2\,(m\,s+1)}-\left((m\,s+1)\left(\varepsilon^2+s\left(-2\,\beta\,\delta\,\varepsilon+i\,\delta^2+\varepsilon^2\left(m+i\,\beta^2\right)\right)\right)+2\,x\,(m\,s+1)\,(\beta\,\varepsilon-i\,\delta)+2\,\beta^2\,x^2\right)\Big/\left(2\,(m\,s+1)\left(m\,s+2\,i\,\beta^2\,s+1\right)\right)\right)$$

```
Timing[FullSimplify[eqn315 /. B → EGT]]
```

{0.203, 0}

```
(* This is intensity,
or the modulus  squared of the Gaussian package under consideration *)
Clear[ABS2EGT];
SetDelayed@@
    {ABS2EGT[x_, s_, δ_, ε_, m_ ], ComplexExpand [Sqrt[2] * Abs[EGT[x, s]] ^ 2] /. β → 1 / Sqrt[2] /.
        Exp[a_] :→ Exp[Collect[#, ε, Factor] & /@ Together[a]]};
TraditionalForm[ABS2EGT[x, s, δ, ε, m ]]
```

$$\exp\left(\left(\varepsilon^2\left(-2\,m^2\,s^2-4\,m\,s-s^2-2\right)-2\sqrt{2}\,\varepsilon\,(m\,s+1)\,(x-\delta\,s)-2\,(x-\delta\,s)^2\right)\Big/\left(2\left(m^2\,s^2+2\,m\,s+s^2+1\right)\right)\right)\Big/\left(\sqrt{(m\,s+1)^2+s^2}\right)$$

```
{ABS2EGT[x, s, 0, 0, -2], ABS2EGT[0, 1 / 2, 0, 0, -2]}
```

$$\left\{\frac{e^{-\frac{x^2}{1-4s+5s^2}}}{\sqrt{(1-2s)^2+s^2}},\ 2\right\}$$

```
(* Graph of intensity for the given Gaussian mode . *)
Plot3D[ABS2EGT[x, s, 0, 0, -2], {x, -2, 2}, {s, -1, 1.5},
  PlotPoints→20, PlotRange→{0, 2.25}, AxesLabel→{x, s, "Intensity"}]
```




```
(* Levels of intensity from  the previous graph. *)
ContourPlot[ABS2EGT[x, s, 0, 0, -2], {x, -2, 2}, {s, -1, 1.5}, Axes→True, AxesLabel→{x, s}]
```



3.4 Breathing Spiral Laser Beams (Laguerre-Gaussian Modes)

■ Isotropic planar harmonic oscillator in a perpendicular magnetic field

In[45]:=
```
(* This is Equation (3.22). *)
eqn322 = I * D[Ψ[X, Y, T], T] + D[Ψ[X, Y, T], X, X] + D[Ψ[X, Y, T], Y, Y] -
    (X^2 + Y^2) * Ψ[X, Y, T] + I * ω * (Y * D[Ψ[X, Y, T], X] - X * D[Ψ[X, Y, T], Y])
```

Out[45]= $-(X^2+Y^2)\,\Psi[X, Y, T] + i\,\Psi^{(0,0,1)}[X, Y, T] + \Psi^{(0,2,0)}[X, Y, T] + i\,\omega\,(-X\,\Psi^{(0,1,0)}[X, Y, T] + Y\,\Psi^{(1,0,0)}[X, Y, T]) + \Psi^{(2,0,0)}[X, Y, T]$

In[46]:=
```
(* This is Equation (3.23) in Cartesian coordinates. *)
Psi[X_, Y_, T_] := Exp[-(X^2 + Y^2) / 2] *
      LaguerreL[n, m , X^2 + Y^2] * Exp[-I * (4 * n + 2 * (m + 1) - m * ω) * T] * (X + I * Y) ^m ;
TraditionalForm[Psi[X, Y, T]]
```

Out[47]//TraditionalForm=

$$(X+i\,Y)^m\;L_n^m\,(X^2+Y^2)\exp\left(\frac{1}{2}(-X^2-Y^2)-i\,T\,(-m\,\omega+2\,(m+1)+4\,n)\right)$$

```
TableForm [Table[TraditionalForm[Psi[X, Y, T]], {n, 0, 2}]]
```

$(X+i\,Y)^m\;\exp\left(\frac{1}{2}(-X^2-Y^2)-i\,T\,(2\,(m+1)-m\omega)\right)$

$(X+i\,Y)^m\;(m-X^2-Y^2+1)\exp\left(\frac{1}{2}(-X^2-Y^2)-i\,T\,(-m\omega+2\,(m+1)+4)\right)$

$\frac{1}{2}(X+i\,Y)^m\;(m^{\;2}-2\,mX^2-2\,mY^2+3\,m+X^4+2\,X^2\,Y^2-4\,X^2+Y^4-4\,Y^2+2)\exp\left(\frac{1}{2}(-X^2-Y^2)-i\,T\,(-m\omega+2\,(m+1)+8)\right)$

```
Timing[FullSimplify[eqn322 /. Ψ → Psi]]
```

{0.64, 0}

■ Isotropic planar harmonic oscillator

```
FullSimplify[
  Psi[X, Y, T] /. {X → Cos[ω * t] * x + Sin[ω * t] * y, Y → -Sin[ω * t] * x + Cos[ω * t] * y, T → t},
  Element [m , Integers]]
```

$e^{-2\,i\,(1+m\;\;+2\,n)\,t-\frac{x^2}{2}-\frac{y^2}{2}}\,(x+i\,y)^m$ LaguerreL[n, m , x^2+y^2]

```
(* This function does not any more  depend on omega . *)
Psi1[x_, y_, t_] := Evaluate[%];
TraditionalForm[Psi1[x, y, t]]
```

$$(x+i\,y)^m\;L_n^m\,(x^2+y^2)\,e^{-2\,i\,t\,(m\;\;+2\,n+1)-\frac{x^2}{2}-\frac{y^2}{2}}$$

```
Timing[FullSimplify[eqn322 /. ω → 0 /. Ψ → Psi1]]
```

{0.172, 0}

■ Planar oscillator

In[48]:=
```
(* This is Equation (3.24). *)
eqn324 = 2 * I * D[A[x, y, s], s] + D[A[x, y, s], x, x] + D[A[x, y, s], y, y] - (x^2 + y^2) * A[x, y, s]
```

Out[48]= $-(x^2+y^2)\,A[x, y, s] + 2\,i\,A^{(0,0,1)}[x, y, s] + A^{(0,2,0)}[x, y, s] + A^{(2,0,0)}[x, y, s]$

Psi2[x_, y_, s_] := $e^{\frac{1}{2}i(-2\,(1+m\;\;+2\,n)\,s+i\,(x^2+y^2))}\,(x+i\,y)^m$ LaguerreL[n, m , x^2+y^2];
TraditionalForm[Psi2[x, y, s]]

$$(x+i\,y)^m\;L_n^m\,(x^2+y^2)\exp\left(\frac{1}{2}\,i\,(-2\,s\,(m+2\,n+1)+i\,(x^2+y^2))\right)$$

```
Timing[FullSimplify[eqn324 /. A → Psi2]]
```

{0.406, 0}

```
(* Compute  a (Groebner) basis for the ideal of operators that annihilate Psi2. *)
ann = Annihilator[Psi2[x, y, s], {Der[s], Der[x], Der[y]}]
```

$\{y D_x - x D_y + i m, D_s + (i + i m + 2 i n),$
$(x^2 y + y^3) D_y^2 + (-x^2 - 2 i m x y + y^2) D_y + (i m x - m^2 y + 2 y^3 + 2 m y^3 + 4 n y^3 - x^2 y^3 - y^5)\}$

```
(* This proves that Psi2 satisfies equation (3.24). *)
OreReduce[2 * I * Der[s] + Der[x] ^ 2 + Der[y] ^ 2 - x ^ 2 - y ^ 2, ann]
```

0

```
(* This tries to FIND equation (3.24) automatically. *)
FindRelation[ann, Support → {1, Der[s], Der[x] ^ 2, Der[y] ^ 2}]
```

$\{D_s + (i + i m + 2 i n), D_x^2 + D_y^2 + (2 + 2 m + 4 n - x^2 - y^2)\}$

```
(* Equation (3.24) corresponds to a linear combination of the above two operators. *)
(2 I) ** %[[1]] + %[[2]]
```

$D_x^2 + D_y^2 + 2 i D_s + (-x^2 - y^2)$

```
(* Here is another ad hoc trick to FIND
  (3.24) by observing that it is free of m  and n. *)
FindRelation[ann, Eliminate → {m , n}]
```

$\{D_x^2 + D_y^2 + 2 i D_s + (-x^2 - y^2)\}$

```
Simplify[ApplyOreOperator[First[%], A[x, y, s]] == eqn324]
```

True

3.5 Laguerre-Gaussian Beams

```
In[49]:= (* This is Equation (3.26). *)
eqn326 = 2 * I * D[B[x, y, s], s] + D[B[x, y, s], x, x] + D[B[x, y, s], y, y]
```

Out[49]= $2 i B^{(0,0,1)}[x, y, s] + B^{(0,2,0)}[x, y, s] + B^{(2,0,0)}[x, y, s]$

■ Transformation (3.27)

```
(* Plug the substitution (3.27) into equation (3.26): *)
transEqn =
    eqn326 /. B → Function[{x, y, s}, Exp[I * s * (x ^ 2 + y ^ 2) / 2 / (1 + s ^ 2)] / (1 + s ^ 2) ^ (1 / 2) *
          A[x / Sqrt[1 + s ^ 2], y / Sqrt[1 + s ^ 2], ArcTan[s]]];
```

```
(* Divide by the exponential factor and clear denominators. *)
transEqn = Numerator [Together[transEqn / Exp[I * s * (x ^ 2 + y ^ 2) / 2 / (1 + s ^ 2)]]]
```

$$-x^2 A\left[\frac{x}{\sqrt{1+s^2}}, \frac{y}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] - y^2 A\left[\frac{x}{\sqrt{1+s^2}}, \frac{y}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] +$$
$$2 i A^{(0,0,1)}\left[\frac{x}{\sqrt{1+s^2}}, \frac{y}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] + 2 i s^2 A^{(0,0,1)}\left[\frac{x}{\sqrt{1+s^2}}, \frac{y}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] +$$
$$A^{(0,2,0)}\left[\frac{x}{\sqrt{1+s^2}}, \frac{y}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] + s^2 A^{(0,2,0)}\left[\frac{x}{\sqrt{1+s^2}}, \frac{y}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] +$$
$$A^{(2,0,0)}\left[\frac{x}{\sqrt{1+s^2}}, \frac{y}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right] + s^2 A^{(2,0,0)}\left[\frac{x}{\sqrt{1+s^2}}, \frac{y}{\sqrt{1+s^2}}, \mathrm{ArcTan}[s]\right]$$

```
(* Now substitute the variables x, y and s in order to get (3.24). *)
Together[(transEqn /. {x → x * Sqrt[1 + s ^ 2], y → y * Sqrt[1 + s ^ 2], ArcTan[s] → s}) / (1 + s ^ 2)]
```

$-x^2 A[x, y, s] - y^2 A[x, y, s] + 2 i A^{(0,0,1)}[x, y, s] + A^{(0,2,0)}[x, y, s] + A^{(2,0,0)}[x, y, s]$

```
Simplify[% == eqn324 /. t → s]
```

```
True
```

Corresponding solution of equation (3.26) obtained by transformation (3.27)

In[50]:=
```
(* This is one of the Laguerre-Gaussian solution of the parabolic equation (3.26). *)
LG[x_, y_, s_] := e^(-(x^2+y^2)/(2+2 s^2) - i (1+m +2 n) ArcTan[s]) ((x + i y)/Sqrt[1 + s^2])^m LaguerreL[n, m , (x^2 + y^2)/(1 + s^2)] *
     (1 + s^2)^(-1/2) * Exp[I * s * (x^2 + y^2) / (2 * (1 + s^2))];
TraditionalForm[LG[x, y, s]]
```

Out[51]//TraditionalForm=

$$\frac{1}{\sqrt{s^2+1}}\left(\frac{x+i\,y}{\sqrt{s^2+1}}\right)^m L_n^m\left(\frac{x^2+y^2}{s^2+1}\right)\exp\left(-i\,(m+2\,n+1)\tan^{-1}(s)+\frac{i\,s\left(x^2+y^2\right)}{2\left(s^2+1\right)}-\frac{x^2+y^2}{2\,s^2+2}\right)$$

```
Timing[FullSimplify[eqn326 /. B → LG]]
```

```
{2.387, 0}
```

```
(* This computes  a system  of PDEs for LG w.r.t. s, x and y. *)
Timing[ann = Annihilator[LG[x, y, s], {Der[s], Der[x], Der[y]}]]
```

```
{0.234, {y Dx - x Dy + i m ,
    (2 y + 2 s^2 y) Ds + (2 s x^2 + 2 s y^2) Dy + (-2 i m  s x + 2 i y + 2 i m  y + 4 i n y + 2 s y - i x^2 y - i y^3),
    (x^2 y + s^2 x^2 y + y^3 + s^2 y^3) Dy^2 + (-x^2 - s^2 x^2 - 2 i m  x y - 2 i m  s^2 x y + y^2 + s^2 y^2 - 2 i s x^2 y^2 - 2 i s y^4) Dy +
      (i m  x + i m  s^2 x - m ^2 y - m ^2 s^2 y - 2 m  s x y^2 + 2 y^3 + 2 m  y^3 + 4 n y^3 - 2 i s y^3 - x^2 y^3 - y^5)}}
```

```
(* This finds a linear combination of the above operators such that only LG_s,
LG_xx, LG_yy occur. *)
(* The result is exactly Equation (3.26). *)
FindRelation[ann, Support → {Der[s], Der[x]^2, Der[y]^2}]
```

```
{Dx^2 + Dy^2 + 2 i Ds}
```

Multi-parameter Laguerre-Gaussian solution (3.28) of equation (3.26)

In[52]:=
```
(* This is a similar verificationthat the multi -
  parameter  solution (3.28) safisfies the parabolic equation (3.26). *)
MLG[x_, y_, s_] := 1/(1 + 2 s α + I * s β^2)
     e^(-((x^2+y^2) (-2 i α+β^2) + 2 x (-i δ1+β ε1) + 2 y (-i δ2+β ε2) + i s (δ1^2+δ2^2) - 2 s β (δ1 ε1+δ2 ε2) + (1+2 s α) (ε1^2+ε2^2))/(2 (1+2 s α+i s β^2)) - i (m +2 n) ArcTan[s β^2/(1+2 s α)])
     ((β (x + i y - s (δ1 + i δ2)) + (1 + 2 s α) (ε1 + i ε2)) / (Sqrt[(1 + 2 s α)^2 + s^2 β^4]))^m
     LaguerreL[n, m , ((x β - s β δ1 + ε1 + 2 s α ε1)^2 + (y β - s β δ2 + ε2 + 2 s α ε2)^2) / ((1 + 2 s α)^2 + s^2 β^4)];
TraditionalForm[MLG[x, y, s]]
```

Out[53]//TraditionalForm=

$$\frac{1}{2\,\alpha\,s+i\,\beta^2\,s+1}\left(((\varepsilon 1+i\,\varepsilon 2)\,(2\,\alpha\,s+1)+\beta\,(-s\,(\delta 1+i\,\delta 2)+x+i\,y))\Big/\left(\sqrt{\beta^4\,s^2+(2\,\alpha\,s+1)^2}\right)\right)^m$$
$$L_n^m\left(\left((\varepsilon 1+2\,\alpha\,\varepsilon 1\,s-\beta\,\delta 1\,s+\beta\,x)^2+(\varepsilon 2+2\,\alpha\,\varepsilon 2\,s-\beta\,\delta 2\,s+\beta\,y)^2\right)\Big/\left(\beta^4\,s^2+(2\,\alpha\,s+1)^2\right)\right)$$
$$\exp\left(-i\,(m+2\,n)\tan^{-1}\left(\frac{\beta^2\,s}{2\,\alpha\,s+1}\right)-\left(\left(\varepsilon 1^2+\varepsilon 2^2\right)(2\,\alpha\,s+1)-2\,\beta\,s\,(\delta 1\,\varepsilon 1+\delta 2\,\varepsilon 2)+i\,s\left(\delta 1^2+\delta 2^2\right)+\right.\right.$$
$$\left.\left.\left(\beta^2-2\,i\,\alpha\right)\left(x^2+y^2\right)+2\,x\,(\beta\,\varepsilon 1-i\,\delta 1)+2\,y\,(\beta\,\varepsilon 2-i\,\delta 2)\right)\Big/\left(2\left(2\,\alpha\,s+i\,\beta^2\,s+1\right)\right)\right)$$

```
Timing[FullSimplify[eqn326 /. B → MLG]]
```

```
{12.48, 0}
```

```
(* This is a derivation of the parabolic equation (3.26) for the multi -
  parameter  solution (3.28). *)
Timing[ann = Annihilator[MLG[x, y, s], {Der[s], Der[x], Der[y]}]]
```

A very large output was generated. Showing a sample of it.

{5.008, {<<1>>}}

show less | show more | show all | set size limit ...

```
ByteCount[ann]
```

433368

```
(* As before, we find an appropriate linear combination of the above operators. *)
(* Again, the result is Equation (3.26). *)
Timing[FindRelation[ann, Support → {Der[s], Der[x]^2, Der[y]^2}]]
```

$\{6.334, \{D_x^2 + D_y^2 + 2\, i\, D_s\}\}$

Graphical example of 2D self-focusing Gaussian mode in Equation (3.28)

In[54]:=
```
(* This is intensity,
or the modulus squared of the Gaussian mode n=m =0 in Equation (3.28). *)
ABS2MG[x_, y_, s_, α_, β_, δ1_, δ2_, ε1_, ε2_] :=
```

$$\frac{1}{(1+2\,s\,\alpha)^2+s^2\,\beta^4} * \mathrm{Exp}\Big[-\big(x^2\,\beta^2+y^2\,\beta^2+s^2\,\beta^2\,\delta1^2-2\,s\,y\,\beta^2\,\delta2+s^2\,\beta^2\,\delta2^2-2\,s\,\beta\,\delta1\,\varepsilon1-4\,s^2\,\alpha\,\beta\,\delta1\,\varepsilon1+\varepsilon1^2+4\,s\,\alpha\,\varepsilon1^2+4\,s^2\,\alpha^2\,\varepsilon1^2+2\,x\,\beta\,(-s\,\beta\,\delta1+\varepsilon1+2\,s\,\alpha\,\varepsilon1)+2\,(1+2\,s\,\alpha)\,\beta\,(y-s\,\delta2)\,\varepsilon2+(\varepsilon2+2\,s\,\alpha\,\varepsilon2)^2\big)\Big/\big((1+2\,s\,\alpha)^2+s^2\,\beta^4\big)\Big]$$

```
TraditionalForm[ABS2MG[x, y, s, α, β, δ1, δ2, ε1, ε2]]
```

Out[55]//TraditionalForm=

$$\frac{1}{\beta^4\,s^2+(2\,\alpha\,s+1)^2}\exp\left(\frac{1}{\beta^4\,s^2+(2\,\alpha\,s+1)^2}\left(-\varepsilon1^2-4\,\alpha^2\,\varepsilon1^2\,s^2+4\,\alpha\,\beta\,\delta1\,\varepsilon1\,s^2-\beta^2\,\delta1^2\,s^2-\beta^2\,\delta2^2\,s^2-4\,\alpha\,\varepsilon1^2\,s-(\varepsilon2+2\,\alpha\,\varepsilon2\,s)^2+2\,\beta\,\delta1\,\varepsilon1\,s-2\,\beta\,x\,(\varepsilon1+2\,\alpha\,\varepsilon1\,s+\beta\,\delta1\,(-s))-2\,\beta\,\varepsilon2\,(2\,\alpha\,s+1)\,(y-\delta2\,s)+2\,\beta^2\,\delta2\,s\,y-\beta^2\,x^2-\beta^2\,y^2\right)\right)$$

```
(* Checking the location of the maximum . *)
{D[ABS2MG[x, y, s, α, β, δ1, δ2, ε1, ε2], x], D[ABS2MG[x, y, s, α, β, δ1, δ2, ε1, ε2], y],
   D[ABS2MG[x, y, s, α, β, δ1, δ2, ε1, ε2], s]};
Simplify[%];
```

$$\%\,/.\left\{x\to-\frac{2\,\alpha\,\delta1+\beta^3\,\varepsilon1}{4\,\alpha^2+\beta^4},\ y\to-\frac{2\,\alpha\,\delta2+\beta^3\,\varepsilon2}{4\,\alpha^2+\beta^4},\ s\to-\frac{2\,\alpha}{4\,\alpha^2+\beta^4}\right\};$$

```
Simplify[%]
```

{0, 0, 0}

```
(* The maximum  is given by *)
```

$$\mathrm{ABS2MG}[x, y, s, \alpha, \beta, \delta1, \delta2, \varepsilon1, \varepsilon2]\,/.\left\{x\to-\frac{2\,\alpha\,\delta1+\beta^3\,\varepsilon1}{4\,\alpha^2+\beta^4},\ y\to-\frac{2\,\alpha\,\delta2+\beta^3\,\varepsilon2}{4\,\alpha^2+\beta^4},\ s\to-\frac{2\,\alpha}{4\,\alpha^2+\beta^4}\right\};$$

```
Simplify[%]
```

$$1+\frac{4\,\alpha^2}{\beta^4}$$

```
(* Choosing special values of the parameters  as follows α=-2,
β=1, δ1=δ2=ε1=ε2=0. *){ABS2MG[x, y, s, -2, 1, 0, 0, 0, 0],
  ABS2MG[0, 0, 0, -2, 1, 0, 0, 0, 0], ABS2MG[0, 0, 4/17, -2, 1, 0, 0, 0, 0]}
```

$$\left\{\frac{e^{\frac{-x^2-y^2}{(1-4s)^2+s^2}}}{(1-4s)^2+s^2}, 1, 17\right\}$$

```
(* Level surface where intensity drops by e. *)
ContourPlot3D[ABS2MG[x, y, s, -2, 1, 0, 0, 0, 0] - 1/e == 0,
  {x, -1, 1}, {y, -1, 1}, {s, -.25, .75}, PlotPoints→5, AxesLabel→{x, y, s}]
```



■ Multi-parameter “elegant” Laguerre-Gaussian modes (3.29)

In[56]:=
```
ELG[x_, y_, s_] :=
```

$$e^{-\frac{2(x^2+y^2)\beta^2+i s(\delta 1^2+\delta 2^2)+2x(-i\delta 1+\beta\varepsilon 1)+2y(-i\delta 2+\beta\varepsilon 2)-2s\beta(\delta 1\,\varepsilon 1+\delta 2\,\varepsilon 2)+(1+i s\beta^2)(\varepsilon 1^2+\varepsilon 2^2)}{2(1+2 i s\beta^2)}}$$

$$(1+2 i s\beta^2)^{-1-m\ -n}(\beta(x+i y-s(\delta 1+i\delta 2))+(1+i s\beta^2)(\varepsilon 1+i\varepsilon 2))^m$$

```
    LaguerreL[n, m , ((xβ - sβδ1 + ε1 + i s β² ε1)² + (yβ - sβδ2 + ε2 + i s β² ε2)²) / (1 + 2 i s β²)];
TraditionalForm[ELG[x, y, s]]
```

Out[57]//TraditionalForm=

$$(1+2 i \beta^2 s)^{-m\ -n-1} \exp(-(((\varepsilon 1^2+\varepsilon 2^2)(1+i\beta^2 s)-2\beta s(\delta 1\,\varepsilon 1+\delta 2\,\varepsilon 2)+i s(\delta 1^2+\delta 2^2)+2\beta^2(x^2+y^2)+2x(\beta\varepsilon 1-i\delta 1)+2y(\beta\varepsilon 2-i\delta 2))/ (2(1+2 i\beta^2 s)))) ((\varepsilon 1+i\varepsilon 2)(1+i\beta^2 s)+\beta(-s(\delta 1+i\delta 2)+x+i y))^m \, L_n^m\left(((\varepsilon 1+i\beta^2\varepsilon 1\, s-\beta\delta 1\, s+\beta x)^2+(\varepsilon 2+i\beta^2\varepsilon 2\, s-\beta\delta 2\, s+\beta y)^2)/(1+2 i\beta^2 s)\right)$$

```
Timing[FullSimplify[eqn326 /. B→ELG]]
```

{5.273, 0}

```
(* This is a derivation of the parabolic equation (3.26) for the multi -
  parameter  solution (3.29). *)
Timing[ann = Annihilator[ELG[x, y, s], {Der[s], Der[x], Der[y]}];]
```

{2.137, Null}

```
ByteCount[ann]
```

229576

```
(* This finds a linear combination of the above equations such that only ELG_s,
ELG_xx and ELG_yy occur. Once again, the result is Equation (3.26). *)
Timing[FindRelation[ann, Support → {Der[s], Der[x]^2, Der[y]^2}]]
```

{8.159, {$D_x^2 + D_y^2 + 2\, i\, D_s$}}

■ Multi-parameter "diffraction free" Laguerre modes (3.30)

In[58]:=
```
DL[x_, y_, s_] :=
   ⅇ^(1/2 (-ⅈ s (δ1^2+δ2^2)-2 x (-ⅈ δ1+β ε1)-2 y (-ⅈ δ2+β ε2)+2 s β (δ1 ε1+δ2 ε2)-(1-ⅈ s β^2) (ε1^2+ε2^2)))
     (β (x + ⅈ y - s (δ1 + ⅈ δ2)) + (1 - ⅈ s β^2) (ε1 + ⅈ ε2))^m  * (1 - 2 ⅈ s β^2)^n
     LaguerreL[n, m , ((x β - s β δ1 + ε1 - ⅈ s β^2 ε1)^2 + (y β - s β δ2 + ε2 - ⅈ s β^2 ε2)^2) / (1 - 2 ⅈ s β^2)];
TraditionalForm[DL[x, y, s]]
```

Out[59]//TraditionalForm=

$$\left(1-2\,i\,\beta^2\,s\right)^n \exp\left(\frac{1}{2}\left(-\left(\varepsilon 1^2+\varepsilon 2^2\right)\left(1-i\,\beta^2\,s\right)+2\,\beta\,s\,(\delta 1\,\varepsilon 1+\delta 2\,\varepsilon 2)-i\,s\left(\delta 1^2+\delta 2^2\right)-2\,x\,(\beta\,\varepsilon 1-i\,\delta 1)-2\,y\,(\beta\,\varepsilon 2-i\,\delta 2)\right)\right)$$
$$\left((\varepsilon 1+i\,\varepsilon 2)\left(1-i\,\beta^2\,s\right)+\beta\,(-s\,(\delta 1+i\,\delta 2)+x+i\,y)\right)^m$$
$$L_n^m\left(\left(\left(\varepsilon 1-i\,\beta^2\,\varepsilon 1\,s-\beta\,\delta 1\,s+\beta\,x\right)^2+\left(\varepsilon 2-i\,\beta^2\,\varepsilon 2\,s-\beta\,\delta 2\,s+\beta\,y\right)^2\right)\Big/\left(1-2\,i\,\beta^2\,s\right)\right)$$

```
Timing[FullSimplify[eqn326 /. B → DL]]
```

{3.837, 0}

```
(* This is a derivation of the parabolic equation (3.26) for the multi -
  parameter  solution (3.30). *)
Timing[ann = Annihilator[DL[x, y, s], {Der[s], Der[x], Der[y]}];]
```

{1.872, Null}

```
ByteCount[ann]
```

228408

```
(* This finds a linear combination of the above equations such that only DL_s,
DL_xx and DL_yy occur. Once again, the result is Equation (3.26). *)
Timing[FindRelation[ann, Support → {Der[s], Der[x]^2, Der[y]^2}]]
```

{8.096, {$D_x^2 + D_y^2 + 2\, i\, D_s$}}

3.6 Bessel-Gaussian Beams

■ Simple Bessel-Gaussian solution of equation (3.26)

In[60]:=
```
BG[x_, y_, s_] := 1/Sqrt[1 + s^2]
   E^((1 - I s)^2 t/(1 + s^2) + I s (x^2 + y^2)/(2 (1 + s^2)) - (x^2 + y^2)/(2 + 2 s^2) - I (1 + m ) ArcTan[s]) ((x + I y)/Sqrt[1 + s^2])^m  Hypergeometric0F1[1 + m , -((1 - I s)^2 t (x^2 + y^2))/(1 + s^2)^2];
TraditionalForm[BG[x, y, s]]
```

Out[61]//TraditionalForm=

$$\frac{1}{\sqrt{s^2+1}}\left(\frac{x+i\,y}{\sqrt{s^2+1}}\right)^m {}_0F_1\!\left(;m+1;-\frac{(1-i\,s)^2\,t\left(x^2+y^2\right)}{\left(s^2+1\right)^2}\right)\exp\!\left(-i\,(m+1)\tan^{-1}(s)+\frac{(1-i\,s)^2\,t}{s^2+1}+\frac{i\,s\left(x^2+y^2\right)}{2\left(s^2+1\right)}-\frac{x^2+y^2}{2\,s^2+2}\right)$$

```
Timing[FullSimplify[eqn326 /. B → BG]]
```

```
{1.935, 0}
```

Derive equation (3.26) from the simple Bessel-Gaussian solution, using the HolonomicFunctions package. As a by-product, this also proves that BG satisfies equation (3.26).

```
Timing[ann = Annihilator[BG[x, y, s], {Der[s], Der[x], Der[y]}]]
```

$$\{0.421, \{y D_x - x D_y + i\,m\,,\ (-2y - 4\,i\,s\,y + 2s^2y)\,D_s + (-2\,i\,x^2 + 2\,s\,x^2 - 2\,i\,y^2 + 2\,s\,y^2)\,D_y + (-2m\,x - 2\,i\,m\,s\,x - 2\,i\,y + 2\,s\,y - 4\,i\,t\,y - i\,x^2y - i\,y^3),$$
$$(-x^2y - 2\,i\,s\,x^2y + s^2x^2y - y^3 - 2\,i\,s\,y^3 + s^2y^3)\,D_y^2 + (x^2 + 2\,i\,s\,x^2 - s^2x^2 + 2\,i\,m\,xy - 4m\,s\,xy - 2\,i\,m\,s^2xy - y^2 - 2\,i\,s\,y^2 + s^2y^2 - 2x^2y^2 - 2\,i\,s\,x^2y^2 - 2y^4 - 2\,i\,s\,y^4)\,D_y + (-i\,m\,x + 2m\,s\,x + i\,m\,s^2x + m^2y + 2\,i\,m^2\,s\,y - m^2s^2y + 2\,i\,m\,xy^2 - 2m\,s\,xy^2 - 2y^3 - 2\,i\,s\,y^3 - 4\,t\,y^3 - x^2y^3 - y^5)\}\}$$

```
(* This finds a linear combination of the above equations such that only BG_s,
BG_xx, BG_yy occur. *)
FindRelation[ann, Support → {Der[s], Der[x]^2, Der[y]^2}]
```

$$\{D_x^2 + D_y^2 + 2\,i\,D_s\}$$

■ Multi-parameter Bessel-Gaussian beams (3.32)

This solution is obtained with the help of (3.31).

In[62]:=
```
MBG[x_, y_, s_] := 1/(1 + 2 s α + I*s*β^2) E^(-((x^2 + y^2) (-2 I α + β^2) + I s (δ1^2 + δ2^2) + 2 x (-I δ1 + β ε1) + 2 y (-I δ2 + β ε2) - 2 s β (δ1 ε1 + δ2 ε2) + (1 + 2 s α) (ε1^2 + ε2^2))/(2 (1 + 2 s α + I s β^2)))
   ((β (x + I y - s (δ1 + I δ2)) + (1 + 2 s α) (ε1 + I ε2)) / (1 + 2 s α + I*s*β^2))^m Hypergeometric0F1[
    m + 1, -((((x β - s β δ1 + ε1 + 2 s α ε1)^2 + (y β - s β δ2 + ε2 + 2 s α ε2)^2) / (1 + 2 s α + I*s*β^2)^2) * t] *
   Exp[t * (1 + 2 s α - I*s*β^2)/(1 + 2 s α + I*s*β^2)];
TraditionalForm[MBG[x, y, s]]
```

Out[63]//TraditionalForm=

$$\frac{1}{2\,\alpha\,s+i\,\beta^2\,s+1}\left(((\epsilon 1+i\,\epsilon 2)\,(2\,\alpha\,s+1)+\beta\,(-s\,(\delta 1+i\,\delta 2)+x+i\,y))/\left(2\,\alpha\,s+i\,\beta^2\,s+1\right)\right)^m$$

$$\exp\!\left(\frac{t\left(2\,\alpha\,s-i\,\beta^2\,s+1\right)}{2\,\alpha\,s+i\,\beta^2\,s+1}-\left(\left(\epsilon 1^2+\epsilon 2^2\right)(2\,\alpha\,s+1)-2\,\beta\,s\,(\delta 1\,\epsilon 1+\delta 2\,\epsilon 2)+i\,s\left(\delta 1^2+\delta 2^2\right)+\left(\beta^2-2\,i\,\alpha\right)\left(x^2+y^2\right)+2\,x\,(\beta\,\epsilon 1-i\,\delta 1)+2\,y\,(\beta\,\epsilon 2-i\,\delta 2)\right)/\left(2\left(2\,\alpha\,s+i\,\beta^2\,s+1\right)\right)\right)$$

$${}_0F_1\!\left(;m+1;-\left(\left(t\left((x\,\beta-s\,\delta 1\,\beta+2\,s\,\alpha\,\epsilon 1+\epsilon 1)^2+(y\,\beta-s\,\delta 2\,\beta+2\,s\,\alpha\,\epsilon 2+\epsilon 2)^2\right)\right)/\left(i\,s\,\beta^2+2\,s\,\alpha+1\right)^2\right)\right)$$

```
Timing[FullSimplify[eqn326 /. B → MBG]]
```

```
{8.299, 0}
```

Again, derive equation (3.26), but now from the solution (3.32), using the HolonomicFunctions package.

```
Timing[ann = Annihilator[MBG[x, y, s], {Der[s], Der[x], Der[y]}]]
```

A very large output was generated. Showing a sample of it.

$$\{11.061, \{(-y\beta + s\beta\delta2 - \varepsilon2 - 2s\alpha\varepsilon2) D_x + (x\beta - s\beta\delta1 + \varepsilon1 + 2s\alpha\varepsilon1) D_y + (-i m\ \beta + i y\beta\delta1 - i x\beta\delta2 - 2 i y\alpha\varepsilon1 - i\delta2\varepsilon1 + 2 i x\alpha\varepsilon2 + i\delta1\varepsilon2), \ll 1 \gg, (x^2 y\beta^3 + y^3\beta^3 + 4 s x^2 y\alpha\beta^3 + 4 s y^3\alpha\beta^3 + 4 \ll 4 \gg \beta^3 + \ll 361 \gg) D_y^2 + \ll 1 \gg + (i m\ x\beta^3 - m^2 y\beta^3 + \ll 1170 \gg + 12 s^2\alpha^2\beta^2\varepsilon2^5 + 8 s^3\alpha^3\beta^2\varepsilon2^5)\}\}$$

show less | show more | show all | set size limit ...

```
Timing[FindRelation[ann, Support → {Der[s], Der[x]^2, Der[y]^2}]]
```

$$\{9.032, \{D_x^2 + D_y^2 + 2 i D_s\}\}$$

■ Multi-parameter elegant Bessel-Gaussian beams (3.33)

In[64]:=

```
EBG[x_, y_, s_] :=
  1/(1 + 2 ⅈ s β^2) ⅇ^(t/(1+2 ⅈ s β^2) - (2 (x^2+y^2) β^2 + ⅈ s (δ1^2+δ2^2) + 2 x (-ⅈ δ1 + β ε1) + 2 y (-ⅈ δ2 + β ε2) - 2 s β (δ1 ε1 + δ2 ε2) + (1 + ⅈ s β^2) (ε1^2 + ε2^2))/(2 (1 + 2 ⅈ s β^2)))
   ((β (x + ⅈ y - s (δ1 + ⅈ δ2)) + (1 + ⅈ s β^2) (ε1 + ⅈ ε2)) / (1 + 2 ⅈ s β^2))^m Hypergeometric0F1[
    1 + m , -((t ((x β - s β δ1 + ε1 + ⅈ s β^2 ε1)^2 + (y β - s β δ2 + ε2 + ⅈ s β^2 ε2)^2)) / (1 + 2 ⅈ s β^2)^2)];
TraditionalForm[
  EBG[
    x,
    y,
    s]]
```

Out[65]//TraditionalForm=

$$\frac{1}{1+2\,i\,\beta^2\,s}\left(\left((\varepsilon1+i\,\varepsilon2)\left(1+i\,\beta^2\,s\right)+\beta\,(-s\,(\delta1+i\,\delta2)+x+i\,y)\right)/\left(1+2\,i\,\beta^2\,s\right)\right)^m$$
$$\exp\left(\frac{t}{1+2\,i\,\beta^2\,s}-\left(\left(\varepsilon1^2+\varepsilon2^2\right)\left(1+i\,\beta^2\,s\right)-2\,\beta\,s\,(\delta1\,\varepsilon1+\delta2\,\varepsilon2)+ i\,s\left(\delta1^2+\delta2^2\right)+2\,\beta^2\left(x^2+y^2\right)+2\,x\,(\beta\,\varepsilon1-i\,\delta1)+2\,y\,(\beta\,\varepsilon2-i\,\delta2)\right)/\left(2\left(1+2\,i\,\beta^2\,s\right)\right)\right)$$
$${}_0F_1\left(;m+1;-\left(\left(t\left(\left(i\,s\,\varepsilon1\,\beta^2+x\,\beta-s\,\delta1\,\beta+\varepsilon1\right)^2+\left(i\,s\,\varepsilon2\,\beta^2+y\,\beta-s\,\delta2\,\beta+\varepsilon2\right)^2\right)\right)/\left(2\,i\,s\,\beta^2+1\right)^2\right)\right)$$

```
Timing[FullSimplify[eqn326 /. B → EBG]]
```

```
{6.864, 0}
```

Again, derive equation (3.26), but now from the solution (3.33), using the HolonomicFunctions package.

```
Timing[ann = Annihilator[EBG[x, y, s], {Der[s], Der[x], Der[y]}];]
```

```
{1417.84, Null}
```

```
ByteCount[ann]
```

```
283904
```

```
Timing[FindRelation[ann, Support → {Der[s], Der[x]^2, Der[y]^2}]]
```

$$\{2.40163, \{D_x^2 + D_y^2 + 2 i D_s\}\}$$

Multi-parameter "diffraction free" Bessel beams (3.34)

In[66]:=
```
DB[x_, y_, s_] :=
  ⅇ^(t (1-2 ⅈ s β^2)+1/2 (-ⅈ s (δ1^2+δ2^2)-2 x (-ⅈ δ1+β ε1)-2 y (-ⅈ δ2+β ε2)+2 s β (δ1 ε1+δ2 ε2)-(1-ⅈ s β^2) (ε1^2+ε2^2)))
   (β (x + ⅈ y - s (δ1 + ⅈ δ2)) + (1 - ⅈ s β^2) (ε1 + ⅈ ε2))^m
   Hypergeometric0F1[1 + m , -t ((x β - s β δ1 + ε1 - ⅈ s β^2 ε1)^2 + (y β - s β δ2 + ε2 - ⅈ s β^2 ε2)^2)];
TraditionalForm[DB[x, y, s]]
```

Out[67]//TraditionalForm=

$$\left((\varepsilon 1 + i\,\varepsilon 2)\left(1 - i\,\beta^2 s\right) + \beta\,(-s\,(\delta 1 + i\,\delta 2) + x + i\,y)\right)^m \exp\Bigg(t\left(1 - 2\,i\,\beta^2 s\right) + \frac{1}{2}\left(-\left(\varepsilon 1^2 + \varepsilon 2^2\right)\left(1 - i\,\beta^2 s\right) + 2\,\beta\,s\,(\delta 1\,\varepsilon 1 + \delta 2\,\varepsilon 2) - i\,s\left(\delta 1^2 + \delta 2^2\right) - 2\,x\,(\beta\,\varepsilon 1 - i\,\delta 1) - 2\,y\,(\beta\,\varepsilon 2 - i\,\delta 2)\right)\Bigg)$$

$$ {}_0F_1\left(; m + 1; -t\left(\left(-i\,s\,\varepsilon 1\,\beta^2 + x\,\beta - s\,\delta 1\,\beta + \varepsilon 1\right)^2 + \left(-i\,s\,\varepsilon 2\,\beta^2 + y\,\beta - s\,\delta 2\,\beta + \varepsilon 2\right)^2\right)\right)$$

```
Timing[FullSimplify[eqn326 /. B → DB]]
```

{3.541, 0}

Again, derive equation (3.26), but now from the solution (3.34), using the HolonomicFunctions package.

```
Timing[ann = Annihilator[DB[x, y, s], {Der[s], Der[x], Der[y]}];]
```

{8.331, Null}

```
Timing[FindRelation[ann, Support → {Der[s], Der[x]^2, Der[y]^2}]]
```

{4.4, $\{D_x^2 + D_y^2 + 2\,i\,D_s\}$}

Graphical examples of "diffraction free" Bessel beams (3.34)

In[68]:=
```
(* This is intensity, or the modulus  squared of the "diffraction-free" Bessel mode  *)
(* for m =0 and t=-1 in Equation (3.34). *)
ABS2DB[x_, y_, s_, α_, β_, δ1_, δ2_, ε1_, ε2_] :=
   Exp[-2 - 2 x β ε1 - ε1^2 - ε2 (2 y β + ε2) + 2 s β (δ1 ε1 + δ2 ε2)] *
    Abs[Hypergeometric0F1[1, ((x β - s β δ1 + ε1 - ⅈ s β^2 ε1)^2 + (y β - s β δ2 + ε2 - ⅈ s β^2 ε2)^2)]]^2;
TraditionalForm[ABS2DB[x, y, s, α, β, δ1, δ2, ε1, ε2]]
```

Out[69]//TraditionalForm=

$$\exp\left(-\varepsilon 1^2 + 2\,\beta\,s\,(\delta 1\,\varepsilon 1 + \delta 2\,\varepsilon 2) - 2\,\beta\,\varepsilon 1\,x - \varepsilon 2\,(\varepsilon 2 + 2\,\beta\,y) - 2\right) \left|{}_0F_1\left(; 1; \left(-i\,s\,\varepsilon 1\,\beta^2 + x\,\beta - s\,\delta 1\,\beta + \varepsilon 1\right)^2 + \left(-i\,s\,\varepsilon 2\,\beta^2 + y\,\beta - s\,\delta 2\,\beta + \varepsilon 2\right)^2\right)\right|^2$$

```
{ABS2DB[x, y, s, α, β, δ1, δ2, ε1, ε2],
  ABS2DB[x, y, s, 0, 1/2, 0, 0, 0, 0], ABS2DB[.1, .1, 0.1, 0, 1/2, 0, 0, 0, 0]}
```

$$\Big\{e^{-2-2\,x\,\beta\,\varepsilon 1-\varepsilon 1^2-\varepsilon 2\,(2\,y\,\beta+\varepsilon 2)+2\,s\,\beta\,(\delta 1\,\varepsilon 1+\delta 2\,\varepsilon 2)}\,\mathrm{Abs}\left[\mathrm{Hypergeometric0F1}\left[1, (x\beta - s\beta\delta 1 + \varepsilon 1 - i\,s\beta^2\varepsilon 1)^2 + (y\beta - s\beta\delta 2 + \varepsilon 2 - i\,s\beta^2\varepsilon 2)^2\right]\right]^2,$$
$$\frac{1}{e^2}\mathrm{Abs}\left[\mathrm{Hypergeometric0F1}\left[1, \frac{x^2}{4} + \frac{y^2}{4}\right]\right]^2, 0.136694\Big\}$$

```
(* Sample diffraction-free level surface. *)
ContourPlot3D[ABS2DB[x, y, s, 0, 1/2, 0, 0, 0, 0] == .15, {x, -.51, .51},
  {y, -.51, .51}, {s, -.1, .1}, PlotPoints→3, AxesLabel→{x, y, s}]
```




```
{ABS2DB[x, y, s, 0, 1/2, .5, .5, 0, 0], ABS2DB[.25, .25, 0, 0, 1/2, .5, .5, 0, 0]}
```

$\left\{0.135335\,\mathrm{Abs}\left[\mathrm{Hypergeometric0F1}\left[1, \left(-0.25\,s+\frac{x}{2}\right)^2+\left(-0.25\,s+\frac{y}{2}\right)^2\right]\right]^2, 0.143994\right\}$

```
(* Sample  "slanted" diffraction-free level surface. *)
ContourPlot3D[ABS2DB[x, y, s, 0, 1/2, .5, .5, 0, 0] == .15, {x, -.51, .51},
  {y, -.51, .51}, {s, -.1, .1}, PlotPoints→3, AxesLabel→{x, y, s}]
```



3.7 Spiral Beams

■ Transformation (3.35)

In[70]:=
```
(* This is Equation (3.36). *)
eqn336 = 2*I*D[C[x, y, s], s] + D[C[x, y, s], x, x] + D[C[x, y, s], y, y] -
    (x^2 + y^2) * C[x, y, s] - 2*I*ω*(x*D[C[x, y, s], y] - y*D[C[x, y, s], x])
```

Out[70]= $-(x^2+y^2)\,C[x,y,s] + 2\,i\,C^{(0,0,1)}[x,y,s] + C^{(0,2,0)}[x,y,s] - 2\,i\,\omega\,(x\,C^{(0,1,0)}[x,y,s] - y\,C^{(1,0,0)}[x,y,s]) + C^{(2,0,0)}[x,y,s]$

```
(* Plug the substitution (3.35) into equation (3.26): *)
transEqn =
    eqn326 /. B→Function[{x, y, s}, Exp[I*s*(x^2 + y^2)/2/(1 + s^2)]/(1 + s^2)^(1/2) *
           C[(Cos[ω*ArcTan[s]]*x + Sin[ω*ArcTan[s]]*y)/Sqrt[1 + s^2],
             (-Sin[ω*ArcTan[s]]*x + Cos[ω*ArcTan[s]]*y)/Sqrt[1 + s^2], ArcTan[s]]];
```

```
(* Divideby the exponential factor and clear denominators . *)
transEqn = FullSimplify[Numerator [Together[transEqn / Exp[I * s * (x^2 + y^2) / 2 / (1 + s^2)]]]]
```

$$-(x^2+y^2)\,C\Big[\frac{1}{\sqrt{1+s^2}}(x\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]+y\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \frac{1}{\sqrt{1+s^2}}(y\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]-x\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \mathrm{ArcTan}[s]\Big]-$$
$$2\,i\sqrt{1+s^2}\,\omega\Big((x\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]+y\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]])\;C^{(0,1,0)}\Big[\frac{1}{\sqrt{1+s^2}}(x\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]+y\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \frac{1}{\sqrt{1+s^2}}(y\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]-x\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \mathrm{ArcTan}[s]\Big]+$$
$$(-y\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]+x\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]])\;C^{(1,0,0)}\Big[\frac{1}{\sqrt{1+s^2}}(x\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]+y\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \frac{1}{\sqrt{1+s^2}}(y\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]-x\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \mathrm{ArcTan}[s]\Big]\Big)+$$
$$(1+s^2)\Big(2\,i\,C^{(0,0,1)}\Big[\frac{1}{\sqrt{1+s^2}}(x\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]+y\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \frac{1}{\sqrt{1+s^2}}(y\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]-x\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \mathrm{ArcTan}[s]\Big]+$$
$$C^{(0,2,0)}\Big[\frac{1}{\sqrt{1+s^2}}(x\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]+y\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \frac{1}{\sqrt{1+s^2}}(y\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]-x\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \mathrm{ArcTan}[s]\Big]+$$
$$C^{(2,0,0)}\Big[\frac{1}{\sqrt{1+s^2}}(x\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]+y\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \frac{1}{\sqrt{1+s^2}}(y\,\mathrm{Cos}[\omega\,\mathrm{ArcTan}[s]]-x\,\mathrm{Sin}[\omega\,\mathrm{ArcTan}[s]]),\ \mathrm{ArcTan}[s]\Big]\Big)$$

```
Simplify[transEqn / (1 + s^2) /.
      {x → Sqrt[1 + s^2] * (x * Cos[ArcTan[s] * ω] - y * Sin[ArcTan[s] * ω]),
       y → Sqrt[1 + s^2] * (x * Sin[ArcTan[s] * ω] + y * Cos[ArcTan[s] * ω])} /.
    ArcTan[s] → s]
```

$$-(x^2+y^2)\,C[x,y,s]+2\,i\,C^{(0,0,1)}[x,y,s]-2\,i\,x\,\omega\,C^{(0,1,0)}[x,y,s]+C^{(0,2,0)}[x,y,s]+2\,i\,y\,\omega\,C^{(1,0,0)}[x,y,s]+C^{(2,0,0)}[x,y,s]$$

```
Simplify[% == eqn336]
```

True

■ Solution (3.37)

In[71]:=
```
(* This is solution (3.37) of Equation (3.26) in the laboratory frame  of reference. *)
```

$$\mathrm{Psi0}[x\_,y\_,s\_]:=\frac{1}{\sqrt{1+s^2}}e^{\frac{1}{2}i\left(\frac{x^2+y^2}{-i+s}-2(1+m+2n)\mathrm{ArcTan}[s]\right)}\left(\frac{x+i\,y}{\sqrt{1+s^2}}\right)^m*\mathrm{LaguerreL}\Big[n,m,\frac{x^2+y^2}{1+s^2}\Big];$$

```
TraditionalForm[Psi0[x, y, s]]
```

Out[72]//TraditionalForm=

$$\frac{1}{\sqrt{s^2+1}}\left(\frac{x+i\,y}{\sqrt{s^2+1}}\right)^m L_n^m\left(\frac{x^2+y^2}{s^2+1}\right)\exp\left(\frac{1}{2}\,i\left(-2\,(m+2\,n+1)\tan^{-1}(s)+\frac{x^2+y^2}{s-i}\right)\right)$$

```
Timing[FullSimplify[eqn326 /. B → Psi0]]
```

```
{2.324, 0}
```

■ Identity (3.42)

In[73]:=
```
(* This is Equation (3.41) in Cartesian coordinates. *)
(* sgn is either +1 or -1. *)
PsiLG0[X_, Y_, T_] :=
    Exp[-I * k * T] * (X + ⅈ sgn Y) ^m * Exp[-(X^2 + Y^2) / 2] * LaguerreL[n, m , X^2 + Y^2];
TraditionalForm[PsiLG0[X, Y, T]]
```

Out[74]//TraditionalForm=

$$e^{\frac{1}{2}(-X^2-Y^2)-i\,k\,T}\,(X+i\,\mathrm{sgn}\,Y)^m\;L_n^m\left(X^2+Y^2\right)$$

In[75]:=
```
(* This is Equation (3.42). *)
eqn342 = eqn336 - 2 * (k + sgn m  ω - m  - 2 n - 1) * C[x, y, s] /. {x → X, y → Y, s → T}
```

Out[75]=

$$-(X^2+Y^2)\,C[X,Y,T]-2\,(-1+k-m-2n+m\ \mathrm{sgn}\,\omega)\,C[X,Y,T]+2\,i\,C^{(0,0,1)}[X,Y,T]+C^{(0,2,0)}[X,Y,T]-2\,i\,\omega\,(X\,C^{(0,1,0)}[X,Y,T]-Y\,C^{(1,0,0)}[X,Y,T])+C^{(2,0,0)}[X,Y,T]$$

```
(* Plug (3.41) into Equation (3.42). *)
Timing[FullSimplify[eqn342 /. C → PsiLG0 /. {{sgn → -1}, {sgn → 1}}]]
```

```
{0.686, {0, 0}}
```

Now derive Equation (3.42) with the HolonomicFunctions package. We do both cases (sgn=1 and sgn=-1) separately.

```
ann = Annihilator[PsiLG0[X, Y, T] /. sgn → 1, {Der[X], Der[Y], Der[T]}]
```

$$\{D_T+i\,k,\ Y D_X-X D_Y+i\,m,\ (X^2 Y+Y^3)\,D_Y^2+(-X^2-2\,i\,m\,X Y+Y^2)\,D_Y+(i\,m\,X-m^2 Y+2Y^3+2m\,Y^3+4n\,Y^3-X^2Y^3-Y^5)\}$$

```
(* Equation (3.42) in operator notation. *)
ToOrePolynomial[eqn342, C[X, Y, T]] /. sgn → 1
```

$$D_X^2+D_Y^2+2\,i\,Y\omega D_X-2\,i\,X\omega D_Y+2\,i\,D_T+(2-2k+2m+4n-X^2-Y^2-2m\,\omega)$$

```
(* Reduction shows that the above operator is in the ideal ann. *)
OreReduce[%, ann /. sgn → 1]
```

```
0
```

```
(* Find Equation (3.42) in the annihilator ideal. *)
First[FindRelation[ann, Support → {1, Der[X]^2, Der[Y]^2}]] + 2 I * (ann[[1]] + ω ann[[2]])
```

$$D_X^2+D_Y^2+2\,i\,Y\omega D_X-2\,i\,X\omega D_Y+2\,i\,D_T+(2-2k+2m+4n-X^2-Y^2-2m\,\omega)$$

```
(* Compare  this derived operator with the equation given in the paper. *)
Simplify[ApplyOreOperator[%, C[X, Y, T]] == eqn342 /. sgn → 1]
```

```
True
```

```
ann = Annihilator[PsiLG0[X, Y, T] /. sgn → -1, {Der[X], Der[Y], Der[T]}]
```

$$\{D_T+i\,k,\ Y D_X-X D_Y-i\,m,\ (X^2 Y+Y^3)\,D_Y^2+(-X^2+2\,i\,m\,X Y+Y^2)\,D_Y+(-i\,m\,X-m^2 Y+2Y^3+2m\,Y^3+4n\,Y^3-X^2Y^3-Y^5)\}$$

```
(* Equation (3.42) in operator notation. *)
ToOrePolynomial[eqn342, C[X, Y, T]] /. sgn → -1
```

$$D_X^2+D_Y^2+2\,i\,Y\omega D_X-2\,i\,X\omega D_Y+2\,i\,D_T+(2-2k+2m+4n-X^2-Y^2+2m\,\omega)$$

```
(* Reduction shows that the above operator is in the ideal ann. *)
OreReduce[%, ann /. sgn → -1]
```

```
0
```

```
(* Find Equation (3.42) in the annihilator ideal. *)
First[FindRelation[ann, Support→{1, Der[X]^2, Der[Y]^2}]] + 2 I * (ann[[1]] + ω ann[[2]])
```

$D_X^2 + D_Y^2 + 2\,i\,Y\,\omega\,D_X - 2\,i\,X\,\omega\,D_Y + 2\,i\,D_T + (2 - 2k + 2m + 4n - X^2 - Y^2 + 2m\ \omega)$

```
(* Compare this derived operator with the equation given in the paper. *)
Simplify[ApplyOreOperator[%, C[X, Y, T]] == eqn342 /. sgn→-1]
```

True

3.8 “Smart” Lens Design

■ Gaussian beam and its energy density inside the lens

In[76]:=

```
(* This is the Gaussian package from Equation (3.8) for n=0. *)
Clear[MS0];
SetDelayed@@{MS0[x_, t_], Simplify[MS[x, t] /. eqns39to314 /. {n→0, μ→1/β, γ→0, κ→0}]};
TraditionalForm[MS0[x, s]]
```

Out[78]//TraditionalForm=

$$\exp\left(-\frac{1}{2\left(\cos(s)+\left(2\,\alpha+i\,\beta^2\right)\sin(s)\right)}\left(i\left(\cos(s)+\left(2\,\alpha+i\,\beta^2\right)\sin(s)\right)\tan^{-1}\left(\frac{\beta^2\sin(s)}{2\,\alpha\sin(s)+\cos(s)}\right)+2\,\alpha\,\varepsilon^2\sin(s)-2\,\beta\,\delta\,\varepsilon\sin(s)+i\,\delta^2\sin(s)+\cos(s)\left(\varepsilon^2+x^2\left(\beta^2-2\,i\,\alpha\right)\right)+i\,x^2\sin(s)+2\,\beta\,\varepsilon\,x-2\,i\,\delta\,x\right)\right)\Big/\left(\sqrt[4]{\pi}\,\sqrt{\left(\frac{1}{\beta}\left(\sqrt{\left((2\,\alpha\sin(s)+\cos(s))^2+\beta^4\sin^2(s)\right)}\right)\right)}\right)$$

```
(* We verify Equation (3.7). *)
Timing[FullSimplify[eqn37 /. A→MS0]]
```

{6.692, 0}

```
(* Intensity of the beam inside the lens. *)
Clear[ABS2MS0];
SetDelayed@@{ABS2MS0[x_, s_, α_, β_, δ_, ε_],
    FullSimplify[ComplexExpand [Abs[MS0[x, s]]^2], Element [{x, s, α, β, δ, ε}, Reals]]};
TraditionalForm[ABS2MS0[x, s, α, β, δ, ε]]
```

$$\left(|\beta|\,\exp\left(-\left(\left(2\,(\sin(s)\,(2\,\alpha\,\varepsilon-\beta\,\delta)+\varepsilon\cos(s)+\beta\,x)^2\right)\Big/\left(4\,\alpha^2+\beta^4-\left(4\,\alpha^2+\beta^4-1\right)\cos(2\,s)+4\,\alpha\sin(2\,s)+1\right)\right)\right)\right)\Big/\left(\sqrt{\pi}\,\sqrt{(2\,\alpha\sin(s)+\cos(s))^2+\beta^4\sin^2(s)}\right)$$

```
ABS2MS0[0, 0, α, β, δ, ε]
```

$$\frac{e^{-\varepsilon^2}\,\text{Abs}[\beta]}{\sqrt{\pi}}$$

```
(* Looking for value of parameters : plot inside a long lens when α=2,
β=1, and 0<s<Pi=1. *)Plot3D[ABS2MS0[x, s, 2, 1, 0, 0], {x, -1.2, 1.2},
  {s, 0, Pi}, PlotPoints→40, PlotRange→{0, 2.4}, AxesLabel→{x, s, "Intensity"}]
```




```
(* Plot of intensity inside of a shorter lens when α=2, β=1,
and 0<s<2.75=1. *)Plot3D[ABS2MS0[x, s, 2, 1, 0, 0], {x, -1.2, 1.2},
  {s, 0, 2.75}, PlotRange→{0, 2.4}, AxesLabel→{x, s, "Intensity"}]
```



- **Gaussian beam and its energy density outside the lens**

In[79]:=
```
(* This is the Gaussian package for n=0 in Equation (3.17). *)
Clear[HGP0];
SetDelayed@@{HGP0[x_, s_], PowerExpand[HGP[x, s] /. {n→0, γ→0}]};
TraditionalForm[HGP0[x, s]]
```

Out[81]//TraditionalForm=

$$\left(\sqrt{\beta}\ \exp\left(\left(-2\,\varepsilon\,s\,(\alpha\,\varepsilon-\beta\,\delta)+i\left(2\,x\,(\delta+\alpha\,x)-\delta^2\,s\right)-(\varepsilon+\beta\,x)^2\right)\Big/\left(2\left(2\,\alpha\,s+i\,\beta^2\,s+1\right)\right)-\frac{1}{2}\,i\tan^{-1}\left(\frac{\beta^2\,s}{2\,\alpha\,s+1}\right)\right)\right)\Big/ \left(\sqrt[4]{\pi}\ \sqrt[4]{\beta^4\,s^2+(2\,\alpha\,s+1)^2}\right)$$

```
(* We verify Equation (3.15). *)
Timing[FullSimplify[eqn315 /. B→HGP0]]
```

```
{0.219, 0}
```

```
Clear[ABS2HGP0];
SetDelayed@@{ABS2HGP0[x_, s_, α_, β_, δ_, ε_],
      Simplify[ComplexExpand [
          FullSimplify[Abs[HGP0[x, s]]^2, Element [{x, s, α, β, δ, ε}, Reals]]]]};
TraditionalForm[ABS2HGP0[x, s, α, β, δ, ε]]
```

$$\frac{\sqrt{\beta^2}\,\exp\left(-\frac{(\varepsilon+2\,\alpha\,\varepsilon\,s-\beta\,\delta\,s+\beta\,x)^2}{s^2\,(4\,\alpha^2+\beta^4)+4\,\alpha\,s+1}\right)}{\sqrt{\pi}\,\sqrt{s^2\,(4\,\alpha^2+\beta^4)+4\,\alpha\,s+1}}$$

```
ABS2HGP0[0, 0, α, β, δ, ε]
```

$$\frac{e^{-\varepsilon^2}\sqrt{\beta^2}}{\sqrt{\pi}}$$

```
(* Plot of intensity outside the lens when α=2, β=1, and s<0. *)
Plot3D[ABS2HGP0[x, s, 2, 1, 0, 0], {x, -1.2, 1.2}, {s, -.5, 0},
  PlotPoints→20, PlotRange→{0, 2.4}, AxesLabel→{x, s, "Intensity"}]
```




```
(* We verify the continuity at s=0. *)
{HGP0[x, 0], MS0[x, 0]}
FullSimplify[HGP0[x, 0] - MS0[x, 0], β > 0]
```

$$\left\{\frac{e^{\frac{1}{2}(2\,i\,x\,(x\alpha+\delta)-(x\beta+\varepsilon)^2)}\sqrt{\beta}}{\pi^{1/4}},\ \frac{e^{\frac{1}{2}(-x^2\,(-2\,i\,\alpha+\beta^2)+2\,i\,x\,\delta-2\,x\,\beta\,\varepsilon-\varepsilon^2)}}{\pi^{1/4}\sqrt{\frac{1}{\beta}}}\right\}$$

0

```
test = {ABS2MS0[0, s, 2, 1, 0, 0], ABS2HGP0[0, s, α, β, 0, 0]};
Simplify[Last[Solve[Simplify[Equal@@@{test^2, D[test, s]} /. s→11/4], {α, β}]]]
```

$$\left\{\alpha\to-\frac{2\left(99+72\,\mathrm{Cos}\left[\frac{11}{2}\right]-76\,\mathrm{Sin}\left[\frac{11}{2}\right]\right)}{1233+488\,\mathrm{Cos}\left[\frac{11}{2}\right]-1124\,\mathrm{Sin}\left[\frac{11}{2}\right]},\ \beta\to\frac{4}{\sqrt{1233+488\,\mathrm{Cos}\left[\frac{11}{2}\right]-1124\,\mathrm{Sin}\left[\frac{11}{2}\right]}}\right\}$$

```
{α1, β1} = Last /@%;
N[{α1, β1}, 15]
```

{-0.171717930330333, 0.0821326116666958}

```
(* Intensity outside the lens when α=α1, β=β1, s>2.75=l. *)
FullSimplify[ABS2HGP0[x, s, N[α1], N[β1], 0, 0]]
```

$$\frac{0.0463384\, e^{-\frac{0.00674577 x^2}{1+(-0.686872+0.117994 s) s}}}{\sqrt{1+(-0.686872+0.117994 s) s}}$$

```
(* We verify (symbolically !) the continuity of intensity at s=2.75=l. *)
FullSimplify[ABS2MS0[x, 11/4, 2, 1, 0, 0] - ABS2HGP0[x, 11/4, α1, β1, 0, 0]]
```

0

```
(* Plot of intensity outside the lens when s>2.75=l. *)
Plot3D[ABS2HGP0[x, s, α1, β1, 0, 0], {x, -1.2, 1.2},
  {s, 2.75, 3.25}, PlotRange→{0, 2.4}, AxesLabel→{x, s, "Intensity"}]
```



■ Example: Gaussian beam propagation through the lens

```
(* Figure 3. Propagation of the beam  through
  the lens: Combine  the last three graphs together. *)
Plot3D[Which[s < 0, ABS2HGP0[x, s, 2, 1, 0, 0], s < 2.75, ABS2MS0[x, s, 2, 1, 0, 0],
    True, ABS2HGP0[x, s, α1, β1, 0, 0]], {x, -1.2, 1.2}, {s, -0.5, 3.25},
  PlotPoints→50, PlotRange→{0, 2.4}, AxesLabel→{x, s, "Intensity"}]
```



4 Gröbner Basis Methods

```
(* Polynomial ideal generated by the equations of the Ermakov -type system . *)
idealErm = Most[#1 - #2 &@@@ ermakov ] /. x_[t] → x /. c₀ → 1
```

{b + 2 c α + 4 a α² - a β⁴ + α′, c β + 4 a α β + β′, a β² + γ′,
 -f - 2 g α + c δ + 4 a α δ - 2 a β³ ∈ + δ′, -g β + 2 a β δ + ∈′, -g δ + a δ² - a β² ∈² + κ′}

```
(* Eliminate a, b, c, ... *)
Timing[GroebnerBasis[idealErm , {α, β, δ, ∈, α', β', γ', δ', ∈', κ'}, {a, b, c, d, f, g}]]
```

{16.723, {δ² γ′ + β² ∈² γ′ - β δ ∈′ + β² κ′}}

```
(* Try to eliminate the derivatives. There is no such relation. *)
Timing[GroebnerBasis[idealErm , {α, β, δ, ∈, a, b, c, d, f, g}, {α', β', γ', δ', ∈', κ'}]]
```

{741.776, {}}